平成9年11月5日
第３版
プログラマブルハンディキーボード
PC ENTRY-1取扱説明書

目次

はじめに
第１章 使用方法
接続時の注意事項
ＰＣ本体への接続
基本動作の確認

第２章 機能活用
ユーザーサービス機能概要
ユーザーサービスモード
サービスメニューサービス機能選択
サービス機能終了
サービス機能についての注意事項
Ｆ１ キーマクロ設定
Ｆ２ キーコンデータ設定
Ｆ３ ユーザーキャラクタ設定
Ｆ４ 電卓
Ｆ５ パスワード設定
Ｆ６ 拡張ポート設定
Ｆ７ フルキー配列選択
Ｆ８ 環境設定
Ｆ９ リセット
Ｆ１０ バックアップモード
第３章 プログラミング
プログラミング概要
サブルーチンサブルーチン解説
ＢＡＳＩＣサンプルプログラム
第４章 コマンドリファレンス
コマンド一覧表第５章 関係資料
エントリー１ キー番号表
フルキーボード キー番号表
キーコード表ＬＣＤキャラクタコード表
機能コード表あいうえお配列表
使用信号線
Ｉ／Ｏポート表
外形寸法図
仮想画面とは表示画面とはキーコンとは
キーコン・キーマクロ コマンド注意事項
テクニカル・トップページへ戻るはじめに

この度は、エントリー１をご購入いただき、誠にありがとうございます。
エントリー１は、日本電気（株）製 PC9801 のフルキーボードラインに接続できるように開発したプログ
ラマブル・ハンディー・キーボードです。
ＥＮＴＲＹ−１には、ＦＡタイプとＯＡタイプ，ＲＢタイプを用意してございます。
形式名称 タイプ
用途
キー部分
PC ENTRY-1 FA
工場向 メンブレン
PC ENTRY-1 OA 一般事務向 キートップ
PC ENTRY-1 RB 一般事務向 ラバー

各タイプは、キー部分の構成が異なるだけで基本的な性能に相違はありません。

多機能な ＮＥＷ ＰＣ ＥＮＴＲＹ−１ を、ご活用頂き、各種ビジネス及び研究・学習から、趣味・ゲー
ムなどのホビーまで、幅広い用途にお役立て頂ければ幸いです。
＊接続可能な機種につきましては、最新接続確認機器一覧表をご参照ください。
また、各種インタフェースに接続可能な姉妹機もございますので、用途に合わせてご利用ください。

エントリー１の特徴
１．ヒューマンインターフェース
エントリー１は、CPUにHD64180を採用し、ROM32KBとRAM32KBを搭載したキーボードとして、革新的に進歩しました。
PCと人間との接点になるべきキーボードに頭脳を持たせて、新しいヒューマンインターフェースを感じさせる、インテリジェントな入力デバイスです。
２．フル・キーボードの替りに利用する。
１）入力端末として利用する。
電卓サイズのエントリー１は、フルキーボードのように一般のオペレータに威圧感を与えず、誰でも簡単手軽になじめます。
２）目的に合わせてキー配列を変更する。
一般的なデータ入力ソフトでは、テンキーと多少の命令キーで十分です。
エントリー１はオペレートに必要な平均キー数の３０キーです。
３）入出力端末として利用する。
本体上部にＬＣＤを見やすく配したことにより、連続データ入力時の確認など、ディスプレーに目を頻繁に移さなくてもすみます。
４）より高度な機能を活用する。
多数のコマンドを準備し、ＢＡＳＩＣ等から簡単なプログラムでコントロールが自由自在にできます。またユーザーサービス機能を搭載しました。
３．フル・キーボードをインテリジェントにする。
１）データ・コンバート（キーコン機能）
例えばフルキーボードをＪＩＳ配列ではなく、あいうえお配列に変更するというような事が実現できます。またノンシフトも可能。
アプリケーションに即したキーボード配列を自分で設定できます。
２）キーマクロ定型化した業務は、キーマクロの登録でワンタッチ。
各キー最大15文字のキーマクロは、入力処理の簡略化を推進させます。

ご利用に際してŶ 設置場所の注意事項
エントリー１は、精密な電子機器です。
強力な磁界のある場所や、ノイズが発生している場所でご使用になると、誤動作を起こす危険性がありますので、使用場所の周辺環境には十分なご配慮をお願い申し上げます。
Ŷ お取り扱いの注意事項
エントリー１の入力部分はメンブレンスイッチを使用しております。
指以外でのキー操作は、メンブレンスイッチを傷める恐れがありますのでおやめください。
Ŷ 故障や異常の時
エントリー１に故障や異常が認められた時は、直ちに使用を中止し、サポートセンターにご連絡ください。
Ŷ アフターサービスについて
本製品には保証書が添付されておりますので大切に保管してください。
サポートセンター
株式会社テクニカル
ユーザーサポート課
TEL:045-322-4400
FAX:045-322-4416

目次へ戻る第１章 使用方法
Ŷ 接続時の注意事項
1. エントリー１をＰＣのキーボードコネクタに接続する際には、ＰＣ本体の電源が切れている状態で行なってください。
2. キーボードコネクタは、ミニＤＩＮ８ピンのコネクタを使用しておりますのでコネクタに接続する
際には、ピンを折らないよう挿入方向に十分注意してください。
3. コネクタを外す際には、ケーブルを持たずにコネクタ部分を持ってください。
Ŷ ＰＣ本体への接続
ＰＣ本体の電源は切れている状態で、用途に合わせた接続を行なってください。
１．シングル接続（１）フルキーボードのコネクタをＰＣ本体から抜いてください。
（２）エントリー１から出ているコネクタをＰＣ本体のキーボードコネクタに挿入してください。

各種機器への組込やアプリケーションのルーチンワーク等、ＰＣを特定用途で使用する場合に、フルキーボードの替りに簡単に使用できます。
２．シリアル接続
ＮＥＷ ＰＣ ＥＮＴＲＹ−１は、フルキーボードとＰＣ本体の間にシリアル（直列）接続が可能です。
（１）フルキーボードのコネクタを、ＰＣ本体から抜いてください。
（２）エントリー１から出ているコネクタを、ＰＣ本体のキーボードコネクタに挿入してください。
（３）フルキーボードのコネクタを、エントリー１上部側面にあるコネクタに挿入してください。

プログラムの開発からワープロ、作表計算等のアプリケーションソフトまでＰＣを利用する用途に、インテリジェントなキーボードとして、幅広くご活用頂けます。
３．パラレル接続
従来のＥＮＴＲＹ−１は、ツインソケットを利用したパラレル接続方式を採用致しておりましたが、フルキーボードをよりインテリジェント化させて、活用するためにもシリアル接続を推奨致します。（ツインソケットはオプションです）
Ŷ 基本動作の確認
シリアル接続が完了した状態で、エントリー１の基本動作確認を行います。
＜ご注意＞ＰＣ本体の電源が投入されている状態では、コネクターの抜き差しは絶対に行わないでください。故障の原因になります。
１．ＰＣ本体の電源を入れてください。
ＲＯＭモードＢＡＳＩＣを立ち上げてください。
＜参 照＞ＰＣに付属するマニュアルをご参照ください。
２．エントリー１オープニングメッセージが表示されます。
下記のようなオープニングメッセージがエントリー１上部のＬＣＤに表示されます。
ＥＮＴＲＹ−１ Ｖｅｒ×．××
Ｔｅｃｈｎｉｃａｌ Ｃｏｒｐ．

表示内の×．××は、現在のエントリー１ＲＯＭバージョンが表示されます。
また、本体からのリセット信号を表示する場合もあります。
３．エントリー１のキー「５」を押下してください。
ＬＣＤ表示には、入力キーデータがエコーバックされて下記のような表示になります。
５ＮＴＲＹ−１ Ｖｅｒ×．××
Ｔｅｃｈｎｉｃａｌ Ｃｏｒｐ．
４．フルキーボードのリターンキーを押下してください。
入力が受付けられて、ＲＯＭモードＢＡＳＩＣが起動されたことが確認できれば基本動作確認は終了です。

ＰＣ本体からエントリー１への、コマンド動作確認を行う場合、第３章のＢＡＳＩＣサンプルプログラ
ムを使用すると、簡単に確認が行なえます。
目次へ戻る第２章 機能活用
エントリー１の機能を活用するには、次の２つの方法があります。
１．ユーザーサービス機能を利用する。
２．ＰＣよりプログラムでコントロールする。
本章では、ユーザーサービス機能について解説します。
またＰＣより、プログラムでコントロールを行なう場合は第３章のプログラミングをご覧ください。
Ŷ ユーザーサービス機能概要
ユーザーサービス機能とは、エントリー１のヘルプキーを押下することにより、エ ントリー１自身がユーザーに提供できる特別のサービスで、次の９種類があります。
１．キーマクロ設定
２．キーコン設定
３．ユーザーキャラクタ設定
４．電卓
５．パスワード設定
６．拡張ポート設定
７．フルキー配列選択
８．環境設定９．リセット
１０．バックアップモード
ヘルプキーを通常キーとして使用したい場合は、ユーザーサービスモードをＯＦＦにしてください。
Ŷ ユーザーサービスモード
ユーザーサービス機能を利用するには、ユーザーサービスモードをＯＮにしてください。（ＯＮ／ＯＦ
Ｆ状態はバックアップされます。）
ユーザーサービスモード ＯＮ
エントリー１の"1"と"リターン"を同時に押下しながら、ＰＣ本体の電源投入を行ってください。ヘルプキー押下で、ユーザーサービスが提供されます。
ユーザーサービスモード ＯＦＦ
通常のヘルプキーとして利用するには、ユーザーサービスの「Ｆ８環境設定」に用意されている「７．サービスモード」を選択してＯＦＦにするか、エントリー１の"0"と"リターン"を同時に押下しながら、ＰＣ本体の電源投入を行って下さい。
＜参照＞ サービスモードの設定＜ご注意＞
バックアップ期間はフル充電時において約10日間です。
（バックアップ期間は、目安であり保証期間ではありません。）
ユーザーサービスモードがONの状態でのオープニングメッセージです。
ＥＮＴＲＹ−１ Ｖｅｒ×．××
ユーザー サービス モード

Ŷサービスメニュー
エントリー１上のヘルプキーを押下します。
ＬＣＤには次のように、提供できるユーザーサービスメニューの一部が表示されます。
Ｆ・１： キーマクロ セット
Ｆ・２： キーコン セット
下方向の矢印を押下してください。
メニュー画面がスクロール表示され、次のようなユーザーサービスメニューが次々に表示されます。
Ｆ・１： キーマクロ セット
Ｆ・２： キーコン セット
Ｆ・３： ユーサ゛ーキャラクタ
Ｆ・４： テ゛ンタク
Ｆ・５： ハ゜スワート゛
Ｆ・６： ＊＊＊＊＊
Ｆ・７： フルキー ハイレツ
Ｆ・８： カンキョウ セット
Ｆ・９： リセット
Ｆ・１０：ハ゛ックアッフ゜
ＨＥＬＰ：ＥＸＩＴ
上矢印で逆方向にスクロール表示します。
Ŷ サービス機能選択
ユーザーサービス機能は、全部で９つのサービスが準備されており２種類の選択方法があります。
1. ユーザーサービスメニューの画面から直接ファンクションキーで選択します。
2. ĹĻキーでメニューをスクロールさせ、ＬＣＤ左側に出ているカーソルを、受けたいサービスメニュ
ーに合わせて、リターンキーで選択します。
Ŷ サービス機能終了
ユーザーサービス機能を終了したいときは、ＥＸＩＴ（サービスメニューの１番下）を選択するか、もう一度ヘルプキーを押下してください。
＜ご注意＞ Ｆ４・電卓機能の終了は方法が異なります。
＜参 照＞ 電卓機能の終了
Ŷ サービス機能についての注意事項
1. Ｆ６は将来の機能拡張に備えて、システムリザーブとなっていますので、現状ユーザーが利用することはできません。
2. ユーザーサービス機能中で、ヘルプキーに機能を設定することはできません。
ヘルプキーに機能を設定する場合は、ＰＣ本体からのコマンドで設定してください。
3. ユーザーがエントリー１に独自のキー配列を設定した状態で、ユーザーサービスの機能を利用しようとする場合には、キー配列とエコーバックデータの設定に注意してください。
4. ユーザーサービス機能中の各設定は、ユーザーサービス機能を終了した後に有効となります。
目次へ戻る◆F１ キーマクロ設定
フルキーボードからのキーストローク（入力動作）を、マクロとしてエントリー１上のヘルプを除く２９キー個々に登録することが可能です。各キーには、最大６４までのキーストロークが登録可能です。
＜ご注意＞
ヘルプキーは、ユーザーサービス機能の起動に設定されていますので、キーマクロ は設定できません。
登録するとユーザーサービス機能が実行できなくなります。
＜参 照＞ キーコン・キーマクロ コマンド注意事項
Ɣ キーマクロ設定手順
１．特殊キー解除
フルキー入力は、フルキーボードにある以下の特殊キーを全て解除してから始めます。
・ＳＨＩＦＴ
・ＣＡＰＳ
・カナ
・ＧＲＰＨ
・ＣＴＲＬ
２．フルキー入力
サービスメニューからキーマクロセットを選択すると、フルキーボードからのキー入力待ち画面になります。
ＫＥＹ ＩＮＰＵＴ（ｍａｘ４６）＆ ＥＮＴＲＹ ｓｅｔｋｅｙ
この状態で、マクロ登録するキー入力動作をフルキーボードから入力します。
＜ご注意＞・特殊キー押下と解除はそれぞれ１ストロークになります。
・１つのキーマクロに、登録できるストローク数は最大６４ストロークです。
３．キー指定フルキーボードからの入力が終了しましたら、エントリー１上のキーを押下して、キーの割当てを指定します。
＜ご注意＞・設定終了時には、フルキーボードの特殊キーは全て解除されている状態で設定終了にしてください。
．ヘルプキーには指定しないでください。
４．マクロ設定終了（設定続行）
Ｓｅｔ ａｎｏｔｈｅｒ ｋｅｙ？
Ｙｅｓ：１ Ｎｏ：０
Ｙｅｓ：１ 他にマクロ設定するか、設定をもう一度やり直す場合
フルキー入力待ちに戻ります。
Ｎｏ ：０ 設定を終了したい場合

ユーザーサービスメニューに戻ります。
Ɣ キーマクロ設定例
キーマクロ設定の、実行例としてエントリー１のＦ１キーにＭＳ−ＤＯＳコマンドのＤＩＲ（ディレクトリ）を登録してみます。
＜ご注意＞ ＰＣ上ではＯＳとして、ＭＳ−ＤＯＳが動作している
ものとします。
〔ＭＳ−ＤＯＳは米国マイクロソフト社の登録商標です。〕
以下の方法で「Ｄｉｒ ／ｗ RET 」を登録します。
1. フルキーボードの特殊キーを解除してください。
2. HELP を押下して、ユーザーサービス機能のメニューを表示させてください。
3. F１または、キーマクロセットにカーソルを合わせてRETを押下してください。
・フルキーボードからの入力待ち画面になります。
4. フルキーボードのCAPSを押下してください。
・Ｄ を入力する為に英大文字モードにします。
5. Ｄを入力してください。
6. フルキーボードのCAPSを押下してください。
・CAPS を解除して英小文字モードにします。
7. i,r,スペース,／,w,RETと続けて入力してください。
・4),5),6),7)は、登録する文字列の入力です。
8. エントリー１の F１ を押下してください。
・フルキーボードより入力したキーストロークをエントリー１上の F１にセットします。
9. エントリー１の ０を押下してください。
・０で、キーマクロ設定の終了です。
・エントリー１の F1 に Dir/w RET と、登録できました。
・最初のユーザーサービスメニューに戻ります。
10. エントリー１の HELPを押下してください。
・ユーザーサービス機能を終了します。
11. エントリー１の F１を押下し、設定されたマクロを実行させてください。
・ＰＣ上で "Dir /w"が表示され、DIR コマンドが実行されます。
目次へ戻る◆F２ キーコンデータ設定
キーコンとは、キーコードコンバートを意味します。
フルキーボードから入力されたデータを、他のデータにコンバート（変換）してＰＣに出力します。
この機能を利用して、フルキーボードのキー配列を自由に変更することや１つのキー入力を複数のキー
コードに変更することも可能です。
この機能には８種類のトリガー条件があり、特殊キーの押下状態の組合せで１キーにつき８パターンのコンバートが設定可能です。また、１キーのコンバートは、最大１５ストロークの入力が可能です。
例えば・・・
1. カナキーの押下状態の時にのみ、自分で考えたローマ字配列にする。
2. CAPS+SHIFTキーの押下状態の時にのみ、10キーの押下でコマンドを発行する。
など、自由に幅広くお使い頂けます。
＜ご注意＞ 本機能を利用するには、エントリー１がシリアル接続状態にされていることが条件となります。
＜参照＞ シリアル接続
Ɣ トリガー条件一覧
１．ノーマル 特殊キー押下のない状態で有効
２．シフト シフトキーの押下時に有効
３．ＣＡＰＳ ＣＡＰＳキーの押下時に有効
４．ＣＡＰＳ＋シフト ＣＡＰＳ＋シフトキーの押下時に有効
５．カナ カナキーの押下時に有効
６．カナ＋シフト カナ＋シフトキーの押下時に有効
７．ＧＲＰＨ ＧＲＰＨキーの押下時に有効８．ＣＴＲＬ ＣＴＲＬキーの押下時に有効
Ɣ キーコンデータ設定手順
１．特殊キー解除
フルキー入力は、フルキーボードの全特殊キーを解除してから始めてください。
特殊キーは、以下の通りです。
ＳＨＩＦＴ
ＣＡＰＳ
カナ
ＧＲＰH
ＣＴＲＬ
２． トリガー条件選択
トリガー条件の選択をします。
ĹĻキーでメニューをスクロールさせ、数字キーもしくはリターンキーにて選択します。
（選択されたトリガー条件の時に、コンバートされます。）
３．フルキー入力
フルキーボードからのコンバートされる文字のキー入力待ち画面になります。
ＫＥＹ ＩＮＰＵＴ（ｍａｘ１５）＆ ＥＮＴＲＹ ＲＥＴｋｅｙ
この状態で、登録する文字（文字列）をフルキーボードから入力します。

入力が終了したら、エントリー１のリターンキーを押下します。
＜ご注意＞
１．特殊キーの押下と解除は、各々１ストロークになります。
２．最大キー押下数は、ストローク数で文字数ではありません。
３．入力中の特殊キーの押下は、全て解除してから終了してください。
４．キー指定３で登録された文字（文字列）を設定するキー入力待ち画面になります。
ＰＵＳＨ ｓｅｔｋｅｙ（Ｆｕｌｌ ｋｅｙＢｏａｒｄ）
この状態で、フルキーボード上の設定するキーを押下します。
フルキーボード上であれば、どのキーでも設定可能です。
５．設定終了（設定続行）
Ｓｅｔ ａｎｏｔｈｅｒ ｋｅｙ？
Ｙｅｓ：１ Ｎｏ：０
Ｙｅｓ：１ 他のキーコンデータの設定を行なう場合
フルキー入力待ち画面に戻ります。
Ｎｏ ：０ 設定を終了する場合
ユーザーサービスメニューに戻ります。
＜ご注意＞ 本設定は、ユーザーサービス中においては動作しません。
Ɣ キーコンデータの設定例
キーコンデータのトリガー条件としてシフトを選択し、ＥＳＣキーに文字列「ＥＮＴＲＹ−１」を登録してみます。
1. フルキーボードの特殊キーを解除してください。
2. エントリー１のHELPを押下して、ユーザーサービス機能のメニューを表 示させてください。
3. F・2：キーコンセットにカーソルを合せてRETを押下してください。
・トリガー条件の選択画面になります。
4. 2.キーコンシフトにカーソルを合わせてRETを押下してください。
・フルキーボードからの入力待ち画面になります。
5. フルキーボードの CAPSを押下してください。
・英大文字モードにします。

6. ENTRY-1と続けて入力してください。
7. エントリー１のRETを押下してください。
・入力を終了して、セットするキーの入力待ち画面になります。
8. フルキーボードのESCを押下してください。
・「ＥＮＴＲＹ−１」の文字列をESCキーに設定します。
9. エントリー１の０を押下してください。
・０で設定の終了です。
最初のユーザーサービスメニューに戻ります。
10. エントリー１のHELPを押下してください。
・ユーザーサービス機能を終了します。
11. フルキーボード上で SHIFTと同時にESCを押下してみます。（実行）・ＰＣの画面上に「ＥＮＴＲＹ−１」と表示されます。

＊SHIFTが押下されていない状態の場合は、通常のESCキーとして動作します。
目次へ戻る◆F３ ユーザーキャラクタ設定
ＬＣＤキャラクタの一部に、ユーザーが任意のキャラクタを作成し登録することが可能です。キャラク
タは全て点（ドット）で表され、横５×縦８の４０ドット構成です。
Ɣ ユーザーキャラクタの設定手順
１．選択
ユーザーキャラクタは、Ｎｏ．０〜７まで登録することが可能です。
ＬＣＤキャラクタコード表の００Ｈから０７Ｈまで、計８文字がユーザーの自由に定義できる文字に割当てられています。
ＵＳＥＲ ＣＨＲ ＳＥＴ
Ｎｏ．（０−７） ？
ＥＮＴＲＹ−１上の０から７の数字キーで、キャラクタＮｏ．を選択します。
＜参照＞ ＬＣＤキャラクタコード表
２．作成・編集
キャラクタＮｏ．が選択されると、編集画面になります。
＊＊＊・・１ ＣＨＲ Ｎｏ．０
＊・・＊・２ PC
編集画面では、ＬＣＤ左側の ＊ ，・ で、各ドットのＯＮ・ＯＦＦを表し数字１から８で段数を表しています。下段中央には現在編集中のユーザーキャラクタが表示されます。
＊＝ＯＮ
・＝ＯＦＦ
ĹĻĸĺキーでカーソルを移動させ、表示されているキャラクタを見ながら編集して行きます。
ドットのＯＮ・ＯＦＦ切換えは、カーソルを合わせ、リターンキーでトグル式に変わります。
３．設定終了編集が終了したら、ヘルプキーの押下で「登録」「終了」を行い最初のメニュー画面に戻ります。
Ɣドットイメージ
１文字編集画面の全体ドットイメージは次の通りです。（例：PCのイメージ）＜ご注意＞ ８段目は通常アンダーカーソルの表示に使用しますので、キャラクタ作成の際にはご注意ください。
Ɣ ユーザーキャラクタの設定例

ユーザーキャラクタ作成の実行例として、キャラクタNo.１に漢字の「正」を登録し てみます。
以下の方法で「正」を作成・登録します。
1. HELPֺ を押下して、ユーザーサービス機能のメニューを表示させてください。
2. F3または、ユーザーキャラクタにカーソルを合わせて、RETを押下してください。
・キャラクタNo.の選択画面になります。
3. １を押下してください。
・キャラクタNo.１の編集画面が表示されます。
4. ĹĻĺĸやRETで、ドットを次の様に編集してください。

5. HELPで終了します。
・編集したキャラクタを登録し、最初のユーザーサービス
メニューに戻ります。
6. 確認の為に、2. 3.の動作を繰返してください。
・ＣＨＲ No.１には「正」が登録されています。
目次へ戻る◆F４ 電卓
エントリー１を電卓として機能させます。
四則演算、平方根、逆数、メモリ計算が行なえ演算結果をＰＣへ送信することが可能です。有効桁数は
最大８桁です。
１．電卓機能

機能

対応キ
ー

内容

クリア
Ｃ
オールクリ
ＡＣ
ア
¥
平方根
逆数

Ｉ／
Ｘ

ｆ６

１回の押下で直前の入力数値や計算結果をクリアします。
２回の押下で現在までの入力数値や計算結果をオールクリアします。
数値や計算結果の平方根を求めます。

ｆ１０

数値や計算結果の逆数を求めます。

除算

÷

HELP

乗算
減算

×

Ĺ
ĸ

加算
計
少数点
符号反転

−

＋
＝
．

＋
／−

ｆ・５

除算の演算を行います。
被除算数への０の入力は無効になります。
乗算の演算を行います。
減算の演算を行います。

．

加算の演算を行います。
計算結果を表示します。
少数点付きの計算を行います。

，

符号の反転を行います。

ESC

Ｍｉ
ｎ
メモリ読出 ＭＲ

BS

数値や計算結果をメモリに書き込みます。

ĺ

メモリ内容がを読み出します。

メモリ加算 Ｍ＋Ļ

メモリ書込

数値や計算結果をメモリ内容に加算しその結果をメモリに書き込みます。
※メモリ減算は、符号反転の後メモリ加算を行います。

＜ご注意＞（ ）を使用しての、演算の優先順位の変更は出来ません。
２．演算結果のＰＣへの送信モードの選択
F４(MODE)，０ により、送信モードの選択が可能です。
１：送信する
０：送信しない＜初期値＞
ＰＣへの送信モードは「０：送信しない」に設定されています。
３．電卓機能の終了
F４(MODE)，F１ により、電卓機能を終了しユーザーサービスメニュー画面に戻ります。
目次へ戻る◆F５ パスワード設定
パスワードを設定する事により、キー入力（フルキーボードを含む）を禁止する事が可能です。
データ入力中に中座する時など、簡単にキー入力を禁止できます。
工場出荷時には、パスワードチェックを必要としないフリーパスになっています。
Ɣ パスワード設定手順
１．パスワードのセット
ＳＥＴ ＵＰＰＡＳＳＷＯＲＤ ［ ］
エントリー１上のキーで数字５桁を設定してから、ヘルプキーを押下すると設定が終了します。
以降ヘルプキーの押下で本体へのキー出力がロックされ、パスワードの入力待ち（チェック）となります。
２．パスワードのチェック
パスワードが設定された状態においてヘルプキーの押下、もしくはエントリー１電源の再投入が行われた場合に、パスワード入力画面になります。
ＰＣ ＥＮＴＲＹ−１
ＰＡＳＳＷＯＲＤ ［ ］
エントリー１より数字５桁のパスワードを入力してください。
この時には、入力した数字の代わりに＊が表示されます。
正常なパスワードが入力されるとユーザーサービスモードのメインメニューに戻ります。
ここでもう一度ヘルプキーを押下すると、通常の入力状態に戻ります。
（パスワードに誤入力がされた場合には再度入力待ちとなります。）
Ɣ パスワードのパックアップと解除
電源切断後もパスワードは、約10日間バックアップ（保持）されています。またパスワードが設定されたエントリー１は、電源投入時にパスワードチェックから実行されます。
パスワードの解除は、「０００００」を設定してください。
＜ご注意＞パスワードのご利用には、十分注意してください。
バックアップ期間は８時間フル充電した時の目安であり、保証期間ではありません。
温度等の環境条件や製品の老朽化によりバックアップ期間が短くなる場合もありますのでご注意くださ
い。
目次へ戻る

◆F６ 拡張ポート設定
拡張ポートに接続される機器の選択を行い、接続機器のデリミタコードや通信速度等の設定を行います。拡張ポートに機器を接続する場合は、ENTRY-1本体の電源が切れている状態で接続してください。
Ɣ 設定メニュー一覧

１：バーコードリーダー バーコードリーダーを接続する場合に選択します。
２：ソノタ
その他の機器を接続する場合に選択します。

１．バーコードリーダー設定
１−１：デリミタ コード＜ＯＮ＞
ＯＮ ：デリミタコードを付加します。
ＯＦＦ：デリミタコードを付加しません。
ＸＸ ：上下カーソルキーによってデリミタコードを変更できます。（ＸＸＨはキーコードです。）
リターンキーを押下することにより、ＯＮ、ＯＦＦの表示が変わります。設定は、ヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜ご注意＞ デリミタコードをＯＮにした場合、キーコード１CH（リターンキー）が設定されます。
＜参 照＞ フルキーボード キー番号表
２．その他の機器設定
２−１：ツウシン パラメタ
２−２：デリミタ コード

２−１：ツウシン パラメタ
２４００ Ｂ７ ＰＥ Ｓ２
拡張ポートへ接続したバーコードリーダー以外の機器の通信パラメータを設定します。
カーソルを矢印キーにて変更したいパラメータの位置へ移動し、リターンキーを押下してください。
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
パラメータ
設定値 LCD表示
ボーレート
ビット長
パリティ
ストップビット

1200 bps
2400 bps
4800 bps
7 bit
8 bit

1200
2400
4800
B7
B8

NONE
ODD
EVEN
1 bit

PN
PO
PE
S1

2 bit２-２.デリミタコードの設定

S2

２−２：デリミタ コード＜ＯＮ＞
ＯＮ ：デリミタコードを付加します。
ＯＦＦ：デリミタコードを付加しません。
ＸＸ ：上下カーソルキーによってデリミタコードを変更できます。（ＸＸＨはキーコードです。）
リターンキーを押下することにより、ＯＮ、ＯＦＦの表示が変わります。設定は、ヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜ご注意＞ デリミタコードをＯＮにした場合、キーコード１CH（リターンキー）が設定されます。
＜参 照＞ フルキーボード キー番号表
目次へ戻る◆F７ フルキー配列選択
フルキーボードの配列を変更します。また、機能としてノンシフトモードを設定するとＣＴＲＬ＋Ｃ等の複合押下をなくすことが可能です。
Ɣ 設定メニュー一覧

標準JIS配列に設定します。
（キー配列の変換後、標準配列に１：JISハイレツ
配列
戻す場合に使用します。）
２：アイウエオ ハイレツ あいうえお配列に設定します。
キーリピートが無く、シフトキー＋
文字キー等の同時押下動作を３：ノンシフト モード
２ストロークに分け入力するための設定です。
モード
ノンシフトノード時における、４：キー キャンセルタイム キー連続押下を任意の時間禁止にします。
（0.0秒から0.9秒）
２．組合せパターン
キー配列は以下の４パターンの組み合わせで設定が可能です。
１：ＪＩＳ配列
２：あいうえお配列
１：ＪＩＳ配列 ＋ ３：ノンシフトモード
２：あいうえお配列 ＋ ３：ノンシフトモード＜ご注意＞
設定のキー配列は、フルキーボードの特殊キー カナ 及びCAPS 押下状態時のみに有効です。
＜参 照＞ あいうえお配列表
目次へ戻る◆F８ 環境設定
エントリー１の各種動作環境を設定します。
Ɣ 設定メニュー一覧

１：カソウ ガメン 仮想画面サイズを設定します。
２：ヒョウジ ガメン 表示画面サイズを設定します。
３：キーリピート
キーリピートの有無を設定します。
４：エコー バック エコーバックモードを設定します。
５：リターン ショリ リターンキー押下時のカーソル処理を設定します。
６：キークリック
キー押下時クリック音の有無を設定します。
７：サービス モード ヘルプキー押下時のサービス対応を設定します。
各メニューでの設定方法は、１０キーより数値を入力するかリターンキーでＯＮ・ＯＦＦを切換えます。
１．仮想画面サイズの設定
実際の仮想画面サイズを設定します。
１： カイソウガメン
Ｘ＝ Ｙ＝
設定範囲 Ｘ×Ｙ≦２０００
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜参 照＞ 仮想画面サイズ設定
仮想画面とは
２．表示画面サイズの設定
実際の表示画面サイズを設定します。
２： ヒョウジガメン
Ｘ＝ Ｙ＝
設定範囲 Ｘ＝００〜１５
Ｙ＝０〜１
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜参 照＞ 表示画面サイズ設定
表示画面とは
３．キーリピートの設定
エントリー１のキーリピートの有無を設定します。
３： キーリピート＜ＯＮ＞
ＯＮ ：リピートします。
ＯＦＦ：リピートしません。

リターンキーを押下することにより、ＯＮとＯＦＦの表示が交互に変ります。
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜ご注意＞キーリピートの設定は全キー一括でＯＮまたはＯＦＦになります。
＜参 照＞ リピートキー設定
４．エコーバックの設定
エントリー１のキー入力データのエコーバックの有無を設定します。
４： エコーバック＜ＯＮ＞
ＯＮ ：エコーバックします。
ＯＦＦ：エコーバックしません。
リターンキーを押下することにより、ＯＮとＯＦＦの表示が交互に変ります。
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜参 照＞ エコーバックモード
５．リターンキー処理の設定
エントリー１のリターンキー押下時のカーソル処理を設定します。
５： リターンキー ショリ＜ＯＮ＞
ＯＮ ：改行処理します。
ＯＦＦ：処理しません。
リターンキーを押下することにより、ＯＮとＯＦＦの表示が交互に変ります。
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜参 照＞ リターンキー処理モード
６．キークリック音の設定
エントリー１のキー入力時のクリック音の有無を設定します。
６： キークリック＜ＯＮ＞
ＯＮ ：クリック音を出します。
ＯＦＦ：クリック音を出しません。
リターンキーを押下することにより、ＯＮとＯＦＦの表示が交互に変ります。
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜参 照＞ キークリック音長設定
７．サービスモードの設定
ユーザーサービス機能提供の有無を設定します。
７： サービスモード＜ＯＮ＞

ＯＮ ：機能の使用可能状態で立ち上がります。
ＯＦＦ：機能の使用不可状態で立ち上がります。
（ヘルプキーは、通常の扱いとなります。）
リターンキーを押下することにより、ＯＮとＯＦＦの表示が交互に変ります。
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
サービスモードＯＦＦからＯＮに設定変更する場合はヘルプキーを押下した状態で、エントリー１に電源を投入してください。
ユーザーサービスモードＯＮに設定され、立ち上がります。
サービスモードがＯＮで立ち上がった時のみ、エントリー１のヘルプキー入力時に、ユーザーサー
ビス機能が動作します。
＜参 照＞ ユーザーサービス機能概要
目次へ戻る◆F９ リセット
エントリー１全体もしくは、各部分のリセットを実行します。
設定条件が全て初期設定値に戻りますので、リセットの扱いには十分注意して行なってください。
Ɣ 設定メニュー一覧

１：フルキー リセット フルキーボードのリセット
２：フルリセット
エントリー１のフルリセット
LCD関係のリセット
３：LCD リセット
４：キーマクロ リセット キーマクロのリセット
５：キーコン リセット キーコンのリセット
６：ハンテン リセット 特殊キー反転の処理
リセットの実行については、メニューのカーソルを移動させ指定するか、メニュー画面より番号入力で直接指定します。
機能が選択された時点で処理を行い、初期画面に戻ります。
但し、フルリセット実行後は、全てのデータがイニシャライズされ初期画面に戻ります。その他のリセット後は、サービスメニューに戻ります。
Ɣ 反転リセットについて
シリアルもしくはパラレル接続時において、フルキーボードの特殊キーの押下状態と、ＰＣ内部の押下状態が違ってしまう場合があります。
キー押下のデータには、押された状態のＭＡＫＥコードと、離された時のＢＲＥＡＫコードがＰＣ本体に送られます。
特殊キーについては、ＯＮとＯＦＦの状態をＰＣ本体がチェックしていないのであるタイミングでデー
タをとりこぼすと、特殊キーの押下状態が反転してしまいます。例えばＣＡＰＳが押下されていないのに英大文字が出る等です。
このような場合に、反転リセットを利用します。
〔特殊キー反転の処理〕
１．ＳＨＩＦＴ ハンテン
２．ＣＡＰＳ ハンテン
３．カナ ハンテン
４．ＧＲＰＨ ハンテン
５．ＣＴＲＬ ハンテン
押下状態が反転している特殊キーの項目を選択してください。
ヘルプキーで、反転リセットが実行され、リセットのメニューに戻ります。
目次へ戻る◆F１０ バックアップモード
エントリー１電源切断時における、ユーザー設定値のバックアップ（保持）状態を設定します。
エントリー１には、メモリの状態をバックアップするスーパーキャパシタが標準実装されており、ユー
ザー
設定値を約１０日間バックアップします。
Ɣ バックアップモードの選択
バックアップモード＜ＯＮ＞
ＯＮ ：バックアップします。
（電源切断時の状態を保持したまま立ち上がります。）
ＯＦＦ：バックアップしません。
（全てのデータがイニシャライズされた状態で立ち上がります。）
リターンキーを押下することにより、ＯＮとＯＦＦの表示が交互に変ります。
設定はヘルプキーで終了し、サービスモードのメニュー画面に戻ります。
＜ご注意＞
バックアップ期間は８時間フル充電した時の目安であり、保証値ではありません。
温度等の環境条件や製品の老朽化により、バックアップ期間が短くなる場合もありますのでご注意ください。
＜参 照＞ バックアップモード＜初期値＞
初期設定値としてバックアップモードがＯＦＦに設定されています。
目次へ戻る第３章 プログラミングŶ プログラミング概要
エントリー１は、ＰＣ本体のキーボードラインから簡単なコマンドを送ることにより、より高度な活用が可能です。
ＰＣ本体のキーボードラインからエントリー１にデータを送るための手順を、ＢＡＳＩＣサブルーチンとして次に示します。
ユーザー自身で本サブルーチンを参考に、エントリー１コントロールプログラムを自由に作成してください。
Ŷ サブルーチン
ＢＡＳＩＣ言語で記述した、標準的なサブルーチンを示します。
また他言語へ移植を検討される場合は、事項のサブルーチン解説をご覧ください。
サブルーチン名は"＊ＳＥＮＤ．ＤＡＴ"です。
コマンドナンバー、パラメーターを、変数名"ＳＥＮＤ＄"の中に代入し、サブルーチン（＊ＳＥＮＤ．ＤＡＴＡ）をコールします。
ここではサブルーチンの実際使用例として、ＬＣＤ表示をクリアして、”ＨＥＬＬＯ！”と表示させてみます。
＜参 照＞
コマンドの詳細については、第４章のコマンドリファレンスを参照してください。
ＬＣＤ表示クリア
メッセージ表示100 '110 ' LCD ヒョウジヲ クリアシ HELLO! ト ヒョウジスル
120 '
130 SEND$=CHR$(&H1B)+CHR$(&H12)+CHR$(&H0) '１２ｈ ＬＣＤ表示クリア
140 GOSUB *SEND.DATA
150 SEND$=CHR$(&H1B)+CHR$(&H1C)+CHR$(&H6)+"HELLO!" '１ｃｈ メッセージ表示
160 GOSUB *SEND.DATA
170 '180 END
190 '200 '-------------------------------------------------210 ' サフ゛ル−チン
220 '-------------------------------------------------230 *SEND.DATA
240 OUT &H2,(&H2 OR INP(&H2)) : DEF SEG=0250 SP.TABLE.1 =PEEK(&H538) : SP.TABLE.2 =PEEK(&H53A)
260 TB.OFFSET.1=PEEK(&H522) : TB.OFFSET.2=PEEK(&H523)
270 FOR I.SEND=1 TO LEN(SEND$)
280 OUT &H43,&H17:OUT &H41,ASC(MID$(SEND$,I.SEND,1)): A%=INP(&H41)
290 FOR J.SEND=1 TO 36: NEXT J.SEND
300 NEXT I.SEND
310 OUT &H43,&H17
320 POKE &H538,SP.TABLE.1 : POKE &H53A,SP.TABLE.2
330 POKE &H522,TB.OFFSET.1 : POKE &H523,TB.OFFSET.2
340 A%=INP(&H41): A%=INP(&H41)
350 OUT &H2,(&HFD AND INP(&H2))
360 WHILE INKEY$<>"" : WEND370 RETURN
380 '-------------------------------------------------Ŷ サブルーチン解説
１．ＯＵＴ ＆Ｈ２，（＆Ｈ２ ＯＲ ＩＮＰ（＆Ｈ２））
以後の処理中はキーボードからの割込みを禁止します。
Ｉ／Ｏポートアドレス＆Ｈ２はＩＭＲ（割込みマスクレジスタ）となっていて、マスクすべき割込
みラインのビットを格納しています。
ＩＭＲのセットされているビットに対応したＩＲ入力の割り込み要求は保留されます。
２．ＤＥＦ ＳＥＧ＝０
セグメントのベースアドレスをシステム共通域に設定します。
３．ＳＰ．ＴＡＢＬＥ．１ ＝ＰＥＥＫ（＆Ｈ５３８）
ＳＰ．ＴＡＢＬＥ．２ ＝ＰＥＥＫ（＆Ｈ５３Ａ）
ＴＢ．ＯＦＦＳＥＴ．１ ＝ＰＥＥＫ（＆Ｈ５２２）
ＴＢ．ＯＦＦＳＥＴ．２ ＝ＰＥＥＫ（＆Ｈ５２３）
ワークエリアの一部を保護します。
４．ＯＵＴ ＆Ｈ４３，＆Ｈ１７
ＰＣ本体側のＵＳＡＲＴ（８２５１Ａ シリアルデータ送受信用ＩＣ）を、送信可の状態にします。
＜参 照＞ Ｉ／Ｏポート表
５．ＯＵＴ ＆Ｈ４１，ＡＳＣ （ＭＩＤ＄（ＳＥＮＤ＄，Ｉ．ＳＥＮＤ，１））
エントリー１へデータを送信します。
実際使用例では、コマンドやメッセージ表示データを、文字列にして、サブルーチンを呼んでいます。
そしてサブルーチン内では、その文字列を１文字づつ出力しています。
６．Ａ％＝ＩＮＰ（＆Ｈ４１）
１文字出力した後に、ダミーデータを取込みます。
＜参 照＞ Ｉ／Ｏポート表
７．ＦＯＲ Ｊ．ＳＥＮＤ＝１ ＴＯ ３６：ＮＥＸＴ Ｊ．ＳＥＮＤ
ＰＣ本体からコマンドや、表示データ列をエントリー１に送出するにあたりデータの送出タイミン
グをとります。
＜ご注意＞
エントリー１の、ＬＣＤ表示やコマンドが正常に動作することが確認できればループの回数をユー
ザー自身で増減してください。
（プログラム内では３６回）また、ＰＣの機種により適切な値（ループ回数値）を使用してください。
８．ＰＯＫＥ ＆Ｈ５３８，ＳＰ．ＴＡＢＬＥ．１
ＰＯＫＥ ＆Ｈ５３Ａ，ＳＰ．ＴＡＢＬＥ．２
ＰＯＫＥ ＆Ｈ５２２，ＴＢ．ＯＦＦＳＥＴ．１
ＰＯＫＥ ＆Ｈ５２３，ＴＢ．ＯＦＦＳＥＴ．２
ワークエリアの一部保護した所を書き戻します。
９．ＯＵＴ ＆Ｈ２，（＆ＨＦＤ ＡＮＤ ＩＮＰ（＆Ｈ２））
キーボードからの割込み禁止を解除します。
10．ＷＨＩＬＥ ＩＮＫＥＹ＄＜＞””：ＷＥＮＤＢＡＳＩＣのキーバッファをクリアーします。
Ŷ ＢＡＳＩＣサンプルプログラム
ＢＡＳＩＣ言語で記述されている本サンプルプログラムは、エントリー１のコマンドをメニューの中から選ぶことにより、全てのコマンドを試せます。
コマンドの詳細については、第４章のコマンドリファレンスを参照してください。
サンプルプログラムはこちら

第４章 コマンドリファレンス
区分
コマンドNO.
名称
汎用コマンド００H
エントリー１ イニシャライズ
０１H
ブザー発鳴
０２H
キークリック音長設定
０３H
バックアップモード
０４H
ヘッダー設定０５H
パケット転送モード
０６H
デリミタ転送モード
０７H
ディレイタイム設定
ＬＣＤコマンド
１０H
ＬＣＤ イニシャライズ
１１H
ＬＣＤ 表示状態設定
１２H
ＬＣＤ 表示クリア
１３H
カーソル種類設定
１４H
カーソルトレースモード
１５H
エコーバックモード
１６H
エコーバックデータ変更
ＬＣＤコマンド
１７H
リターンキー処理モード
１８H
仮想画面サイズ設定
１９H
表示画面サイズ設定
１ＡH
カーソルロケート
１ＢH
表示画面ロケート
１ＣH
メッセージ表示１ＤH
ユーザーキャラクタ設定
キーコマンド２０H
キーイニシャライズ
２１H
キーコンデータ設定
２２H
キーマクロ設定
２３H
キーマスクモード
２４H
フルキー配列選択
２５H
リピートキー設定
２６H
キーリピートインターバル
目次へ戻る第５章 関係資料◆エントリー１ キー番号表

目次へ戻る◆フルキーボード キー番号

目次へ戻る◆キーコード表＜ご注意＞ＰＣ9800シリーズの機種により一部異なる場合があります。
本表は、キーが押された時に出力されるMAKEコード表です。
キーが離された時に出力されるBRAKEコードは最上位ビットが１になります。
目次へ戻る◆LCDキャラクタコード表

キャラクタコードと文字フォントとの対応表＜ご注意＞
００H 〜１ＦH 、８０H 〜９ＦH までは機能コードとなっています。
＜ 参 照＞ 機能コード表
目次へ戻る◆機能コード表

ＬＣＤキャラクタコードの００Hから１ＦHまでと、８０Hから９ＦHまでは機能コードという設定になっており、ＬＣＤエコーバックデータにこのコードを設定するとあらかじめ設定されている機能が動作します。
代表的な機能コードは１ＢHで、ＬＣＤエコーバックデータにこの機能コードが設定されていると、ユー
ザーサービスが起動します。
コード
機能
コード
機能
００H ユーザーキャラクタ
０８H バックスペース
０１H ユーザーキャラクタ
０９H 水平タブ
０２H ユーザーキャラクタ
０ＡH リザーブ
０３H ユーザーキャラクタ
０ＢH リザーブ
０４H ユーザーキャラクタ
０ＣH リザーブ
０５H ユーザーキャラクタ
０ＤH リザーブ
０６H ユーザーキャラクタ
０ＥH リザーブ
０７H ユーザーキャラクタ
０ＦH リザーブ
コード
機能
コード
機能
１０H １文字デリート
１８H リザーブ
１１H カーソル右側デリート １９H カーソルＯＮ
１２H カーソル左側デリート １ＡH カーソルＯＦＦ
１３H １文字インサート
１ＢH ユーザーサービスモード
１４H ＬＣＤクリア
１ＣH リザーブ
１５H ブリンクカーソルＯＮ １ＤH ＬＣＤ表示ＯＮ
１６H ブリンクカーソルＯＦＦ １ＥH ＬＣＤ表示ＯＦＦ
１７H リザーブ
１ＦH リザーブ
コード
機能
コード
機能
８０H カーソルホーム
８８H リザーブ
８１H カーソル １Ｕｐ
８９H リザーブ
８２H カーソル １Ｄｏｗｎ
８ＡH リザーブ
８３H カーソル １Ｒｉｇｈｔ ８ＢH リザーブ
８４H カーソル １Ｌｅｆｔ
８ＣH リザーブ
８５H リザーブ
８ＤH リザーブ
８６H リザーブ
８ＥH リザーブ
８７H リザーブ
８ＦH リザーブ
コード
機能
コード
機能
９０H 表示画面ホーム
９８H リザーブ
９１H 表示画面１Ｕｐ
９９H リザーブ
９２H 表示画面１Ｄｏｗｎ
９ＡH リザーブ
９３H 表示画面１Ｒｉｇｈｔ ９ＢH リザーブ
９４H 表示画面１Ｌｅｆｔ
９ＣH リザーブ
９５H リザーブ
９ＤH リザーブ
９６H リザーブ
９ＥH リザーブ
９７H リザーブ
９ＦH リザーブ
リザーブとは、将来の拡張に備えて予約されているもので、動作が保証されていませんのでエコーバッ
クデータには、設定しないでください。

◆あいうえお配列表

目次へ戻る◆使用信号線

エントリー１の使用している信号線を示します。
端子番号 信号名 フルキーボード エントリー１ ピン・コネクション
１
ＲＳＴ
◯
◯
２
３
４
５
６
７
８

ＧＮＤ
ＲＤＹ
ＲＸＤ
ＲＴＹ
ＮＣ
ＮＣ
＋５Ｖ

◯
◯
◯
◯

◯
◯
◯
◯

◯

◯
（◯印 使用信号線）

目次へ戻る

◆Ｉ／Ｏポート表

Ｉ／Ｏ
ポート
アドレス

コマンド

READ／
WRITE

意味

ＭＯＤＥ
ＭＯＤＥセット
ＷＲＩＴＥ
ＣＯＭＭＡＮＤ 出力
ＣＯＭＭＡＮＤセット
４１ｈ ＤＡＴＡ 入力
ＲＥＡＤ １バイトデータの引取り
４３ｈ ステータス 入力
ＲＥＡＤ ステータスの引取り
４１ｈ ＤＡＴＡ 出力
ＷＲＩＴＥ １バイトデータの送出
４３ｈ

目次へ戻る

◆外形寸法図

目次へ戻る

◆仮想画面とはエントリー１は本体上面に１６文字×２行のＬＣＤ表示器（液晶画面）が付属しています。このＬＣＤは最大３２文字を表示でき、内部ではユーザーが自由に設定できる大きさの画面（仮想画面）を保持し、最大で２０００文字を表示します。
２０００文字以内であれば、ユーザーは自由な縦横サイズを設定することができ、初期状態では１６文
字２行に設定されています。
通常使用する仮想画面サイズの最大値は、Ｘ７９、Ｙ２４となり、ＰＣ等の標準ディスプレイと同等の
８０文字２５行となります。（80×25＝2000文字）

目次へ戻る◆表示画面とは

仮想画面中より、実際にＬＣＤに表示される部分を、表示画面と言います。
例えば、仮想画面サイズが８０文字２５行に設定されているとした場合、表示画面とは、この仮想画面上の一部（１６文字２行分）だけをＬＣＤに表示している、いわば窓みたいな部分を（ウィンドとも言います）指します。
ユーザーが実際に見える部分は、この窓を通した仮想画面の一部と言えます。
そしてこの仮想画面上を、この窓（表示画面）を動かすことにより、必要な部分をユーザーが見ることができるのです。
初期値では表示窓（表示画面）サイズは、最大値（Ｘ１５、Ｙ１）の１６文字２行が設定されており、最小値（Ｘ０、Ｙ０）の１文字１行まで小さくすることができます。

目次へ戻る◆キーコンとはキーコンとは、キーコードコンバータのことです。
例えば、フルキーボードからＡと入力された場合に、Ｂというデータに変換（コンバート）してホスト（ＰＣ本体）にデータを送ることです。また、Ｃと入力した場合に、”ＨＥＬＬＯ”という文字列データにマルチコンバートすることもできます。
コンバートするデータには、条件を設定します。これをトリガー条件と呼び、特殊キー押下状態が８パ
ターンのいずれかに該当する場合に、コンバートを実行します。
〔トリガー条件パターン〕（１）ノーマル（２）シフト（３）ＣＡＰＳ（４）ＣＡＰＳ＋シフト（５）カナ（６）カナ＋シフト（７）ＧＲＰＨ（８）ＣＴＲＬ
例えばトリガー条件を、カナ＋シフトに設定しＡのキーのコンバートデータを”ＨＥＬＬＯ”と設定します。
特殊キーが何も押下されていない状態で、Ａのキーを押すと英小文字の”ａ”がホスト（ＰＣ本体）に出力されます。
また、特殊キーがカナだけ押下されている場合には、カナの”チ”が出力されます。
特殊キーのカナとシフトが押された場合にのみ、”ＨＥＬＬＯ”と出力します。
目次へ戻る◆キーコン・キーマクロ コマンド注意事項

キーコンデータ設定（コマンドNo21）、キーマクロ設定（コマンドNo22）のコマンドについての補足事
項を説明致します。
キーコン、マクロデータ中に、カナ文字、シフト文字等の特殊キーの文字を入れる際には以下の点に注意してください。
キーコード表は、通常フルキーボードの特殊キーオフの状態です。
通常のキーコード指定時には、英字の小文字が指定されており、大文字やカナ文字の表示には、特殊キーのキーコードを指定します。
特殊キーの指定はデータサイズの文字数を必要とします。
例えば
表示データ Ａ チ のデータサイズは６文字です。
データ列は、 70H 1DH F0H 72H 1DH F2H とします。
７０H 特殊キーのシフトキーオン
１ＤH 'Ａ'のキー押下
Ｆ０H 特殊キーのシフトキーオフ
７２H 特殊キーのカナキーオン
１ＤH 'チ'のキー押下Ｆ２H 特殊キーのカナキーオフ
特殊キーを使用するキーコード表示（指定）は特殊キーの動作を入れる必要があります。特殊キー設定の後にキーコードを指定し特殊キーを解除します。
［ブレイクコード表］

目次へ戻る

メイクコード ブレイクコード
SHIFT
F０H
７０H
CAPS
F１H
７１H
F２H
カナ
７２H
GRPH
F３H
７３H
CTRL
F４H
７４H

1010 '/* sample.bas -- Copyright by Technical Industry Japan Corp. */1020 '/***********************************************************************/
1030 '/*
1040 '/**/ＥＮＴＲＹ−１サンプルプログラム

1050 '/**/*/

1060 '/***********************************************************************/
1070 DEFINT I,J,K: ONN=1: OFFF=0: DIM HEX%[256]: COL=ONN
2000 '************************************
2010 ' メインプログラム
2020 '************************************
2030 *MAIN
2040 CONSOLE 0,25,0,1: WIDTH 80,25: CLS: COLOR 72050 IF COL=ONN THEN COLOR 52060 LOCATE 18,0: PRINT "★★★ ＥＮＴＲＹ−１サンプルプログラム ★★★"
2070 CONSOLE 2,232080 RESTORE *MENU.DATA
2090 FOR I=1 TO 34
2100 READ X%,Y%,C%,A$
2110 LOCATE X%,Y%: IF COL=ONN THEN COLOR C%
2120 PRINT A$
2130 NEXT I
2140 CONSOLE 19,62150 WHILE( 1 )2160 EXIT.FLG=OFFF: IF COL=ONN THEN COLOR 72170 WHILE( EXIT.FLG=OFFF )2180

GOSUB *STACK.CLEAR

2190

CLS: INPUT "実行するコマンド番号を入力してください ";MENU$

2200

MENU.NO%=VAL("&H"+MENU$)

2210

IF MENU.NO%=&HE THEN EXIT.FLG=ONN

2220

IF MENU.NO%<&H8 THEN EXIT.FLG=ONN

2230

IF MENU.NO%>&HF AND MENU.NO%<&H1E THEN EXIT.FLG=ONN

2240

IF MENU.NO%>&H1F AND MENU.NO%<&H27 THEN EXIT.FLG=ONN

2250

IF MENU.NO%=&H30 THEN EXIT.FLG=ONN

2260

IF MENU.NO%=&H0 AND MENU$<>"0" THEN EXIT.FLG=OFFF

2270

IF EXIT.FLG=ONN THEN GOSUB *BEEP.CLICK ELSE GOSUB *BEEP.ERROR

2280 WEND
2290 IF MENU.NO%=&HE THEN GOTO *END.PROC
2300 SEND$=CHR$(&H1B)+CHR$(MENU.NO%): PRINT
2500 '************************************
2510 ' 各コマンド処理に分岐する
2520 '************************************
2530 IF MENU.NO%=&H0 THEN GOSUB *INIT.PROC
2540 IF MENU.NO%=&H1 THEN GOSUB *BEEP.PROC
2550 IF MENU.NO%=&H2 THEN GOSUB *CLICK.PROC
2560 IF MENU.NO%=&H3 THEN GOSUB *BACKUP.PROC
2570 IF MENU.NO%=&H4 THEN GOSUB *HEADER.PROC
2580 IF MENU.NO%=&H5 THEN GOSUB *PACKETO.PROC
2590 IF MENU.NO%=&H6 THEN GOSUB *DERIMITA.PROC
2600 IF MENU.NO%=&H7 THEN GOSUB *DLY.PROC
2610 IF MENU.NO%=&H10 THEN GOSUB *LCDINIT.PROC
2620 IF MENU.NO%=&H11 THEN GOSUB *LCD.DSP.PROC
2630 IF MENU.NO%=&H12 THEN GOSUB *LCD.CLS.PROC
2640 IF MENU.NO%=&H13 THEN GOSUB *CURDSP.PROC
2650 IF MENU.NO%=&H14 THEN GOSUB *CURMOV.PROC
2660 IF MENU.NO%=&H15 THEN GOSUB *ECHOBACK.PROC
2670 IF MENU.NO%=&H16 THEN GOSUB *ECHOCHANGE.PROC
2680 IF MENU.NO%=&H17 THEN GOSUB *RETCLEAR.PROC

2690 IF MENU.NO%=&H18 THEN GOSUB *IMAGEAREA.PROC
2700 IF MENU.NO%=&H19 THEN GOSUB *DSPAREA.PROC
2710 IF MENU.NO%=&H1A THEN GOSUB *CURLOCATE.PROC
2720 IF MENU.NO%=&H1B THEN GOSUB *AREALOCATE.PROC
2730 IF MENU.NO%=&H1C THEN GOSUB *DATADSP.PROC
2740 IF MENU.NO%=&H1D THEN GOSUB *USERCHAR.PROC
2750 IF MENU.NO%=&H20 THEN GOSUB *KEYINIT.PROC
2760 IF MENU.NO%=&H21 THEN GOSUB *KEYCON.PROC
2770 IF MENU.NO%=&H22 THEN GOSUB *KEYMACRO.PROC
2780 IF MENU.NO%=&H23 THEN GOSUB *KEYMASK.PROC
2790 IF MENU.NO%=&H24 THEN GOSUB *KEYMODE.PROC
2800 IF MENU.NO%=&H25 THEN GOSUB *KEYREP.PROC
2810 IF MENU.NO%=&H26 THEN GOSUB *KEYTIME.PROC
2820 IF MENU.NO%=&H30 THEN GOSUB *TESTMODE.PROC
2830 IF MENU.NO%<>&H30 THEN GOSUB *SEND.DATA
2840 FOR I.MENU=1 TO 777: NEXT I.MENU
2850 IF COL=ONN THEN COLOR 62860 PRINT "どれかのキーを押して下さい";2870 GOSUB *STACK.CLEAR: A$=INPUT$(1)2880 WEND
3000 '************************************
3010 ' 終了処理を行なう
3020 '************************************
3030 *END.PROC
3040 CONSOLE 0,25,0,1: CLS: CLOSE:
3050 PRINT "お疲れさまでした！！"
3060 FOR I=1 TO 7777: NEXT I
3070 END
10000 '************************************

10010 ' エントリー１イニシャライズｺﾏﾝﾄﾞ処理
10020 '************************************
10030 *INIT.PROC
10040 SEND$=SEND$+CHR$(&H0)
10050 RETURN
10100 '************************************
10110 ' ブザー発鳴ｺﾏﾝﾄﾞ処理
10120 '************************************
10130 *BEEP.PROC
10140 MSG$="ビープ音の時間": MAX%=255: GOSUB *QTY.INP
10150 SEND$=SEND$+CHR$(QTY%)
10160 RETURN
10200 '************************************
10210 ' キークリック音長設定ｺﾏﾝﾄﾞ処理
10220 '************************************
10230 *CLICK.PROC
10240 MSG$="クリック音の時間": MAX%=255: GOSUB *QTY.INP
10250 SEND$=SEND$+CHR$(QTY%)
10260 RETURN
10300 '************************************
10310 ' バックアップモードｺﾏﾝﾄﾞ処理
10320 '************************************
10330 *BACKUP.PROC
10340 INPUT "電源切断時のバックアップを行なうか否か指定して下さい(ON=1,OFF=0) ";BACKUP%
10350 IF BACKUP%>1 THEN GOSUB *BEEP.ERROR: GOTO *BACKUP.PROC
10360 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(BACKUP%)
10370 RETURN
10400 '************************************
10410 ' ヘッダー設定ｺﾏﾝﾄﾞ処理

10420 '************************************
10430 *HEADER.PROC
10440 MSG$="識別子の文字数": MAX%=255: GOSUB *QTY.INP
10450 MAX%=&H7F: MSG$="キーコード": GOSUB *HEX.INP
10460 SEND$=SEND$+CHR$(QTY%)
10470 FOR I=1 TO QTY%
10480 SEND$=SEND$+CHR$(HEX%[I])10490 NEXT I
10500 RETURN
10600 '************************************
10610 ' パケット転送モードｺﾏﾝﾄﾞ処理
10620 '************************************
10630 *PACKETO.PROC
10640 INPUT "動作モードを入力して下さい(ﾌﾙｷｰﾎﾞｰﾄﾞ=1,ENRTY-1=2,両方=3) ";MODE%
10650 IF MODE%>3 THEN GOSUB *BEEP.ERROR: GOTO *PACKETO.PROC
10660 GOSUB *BEEP.CLICK
10670 MSG$="パケットとして溜める文字数": MAX%=&H7F: GOSUB *QTY.INP
10680 SEND$=SEND$+CHR$(MODE%)+CHR$(QTY%)
10690 RETURN
10700 '************************************
10710 ' デリミタ転送モードｺﾏﾝﾄﾞ処理
10720 '************************************
10730 *DERIMITA.PROC
10740 INPUT "動作モードを入力して下さい(ﾌﾙｷｰﾎﾞｰﾄﾞ=1,ENRTY-1=2,両方=3)";MODE%
10750 IF MODE%>3 THEN GOSUB *BEEP.ERROR: GOTO *DERIMITA.PROC
10760 GOSUB *BEEP.CLICK
10770 MAX%=&H7F: QTY%=1: MSG$="指定キーボードのキーコード": GOSUB*HEX.INP
10780 SEND$=SEND$+CHR$(MODE%)+CHR$(HEX%[1])10790 RETURN
10800 '************************************
10810 ' ディレイタイム設定ｺﾏﾝﾄﾞ処理
10820 '************************************
10830 *DLY.PROC
10840 MSG$="ホストへ送信する時の待ち時間": MAX%=255: GOSUB *QTY.INP
10850 SEND$=SEND$+CHR$(QTY%)
10860 RETURN
10900 '************************************
10910 ' ＬＣＤイニシャライズｺﾏﾝﾄﾞ処理
10920 '************************************
10930 *LCDINIT.PROC
10940 SEND$=SEND$+CHR$(&H0)
10950 RETURN
11000 '************************************
11010 ' ＬＣＤ表示状態設定ｺﾏﾝﾄﾞ処理
11020 '************************************
11030 *LCD.DSP.PROC
11040 INPUT "ＬＣＤ表示を行なうか否か指定して下さい(ON=1,OFF=0) ";LCD%
11050 IF LCD%>1 THEN GOSUB *BEEP.ERROR: GOTO *LCD.DSP.PROC
11060 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(LCD%)
11070 RETURN
11100 '************************************
11110 ' ＬＣＤ表示クリアｺﾏﾝﾄﾞ処理
11120 '************************************
11130 *LCD.CLS.PROC
11140 SEND$=SEND$+CHR$(&H0)
11150 RETURN
11200 '************************************

11210 ' カーソル種類設定ｺﾏﾝﾄﾞ処理
11220 '************************************
11230 *CURDSP.PROC
11240 INPUT "カーソルのモードを指定して下さい(ULINE=1,BLINK=2,両方=3) ";CURTYPE%
11250 IF CURTYPE%<1 OR CURTYPE%>3 THEN GOSUB *BEEP.ERROR: GOTO *CURDSP.PROC
11260 *CURDSP.ERROR
11270 INPUT "カーソル表示を行なうか否か指定して下さい(ON=1,OFF=0) ";CUR%
11280 IF CUR%>1 THEN GOSUB *BEEP.ERROR: GOTO *CURDSP.ERROR
11290 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(CURTYPE%)+CHR$(CUR%)
11300 RETURN
11400 '************************************
11410 ' カーソルトレースモードｺﾏﾝﾄﾞ処理
11420 '************************************
11430 *CURMOV.PROC
11440 INPUT "カーソル追従を行なうか否か指定して下さい(ON=1,OFF=0) ";CUR%
11450 IF CUR%>1 THEN GOSUB *BEEP.ERROR: GOTO *CURMOV.PROC
11460 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(CUR%)
11470 RETURN
11500 '************************************
11510 ' エコーバックモードｺﾏﾝﾄﾞ処理
11520 '************************************
11530 *ECHOBACK.PROC
11540 INPUT "エコーバックを行なうか否か指定して下さい(ON=1,OFF=0) ";ECHOBACK%11550 IF ECHOBACK%>1 THEN GOSUB *BEEP.ERROR: GOTO *ECHOBACK.PROC
11560 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(ECHOBACK%)
11570 RETURN
11600 '************************************
11610 ' エコーバックデータ変更ｺﾏﾝﾄﾞ処理
11620 '************************************

11630 *ECHOCHANGE.PROC
11640 QTY%=1: MAX%=&H1D: MSG$="ＥＮＴＲＹ−１のキーナンバー"11650 GOSUB *HEX.INP: EKEY%=HEX%[1]11660 MSG$="エコーデータ数": MAX%=15: GOSUB *QTY.INP
11670 MAX%=&HFF: MSG$="キャラクタコード": GOSUB *HEX.INP
11680 SEND$=SEND$+CHR$(EKEY%)+CHR$(QTY%)
11690 FOR I=1 TO QTY%11692 SEND$=SEND$+CHR$(HEX%[I])11694 NEXT I
11696 RETURN
11700 '************************************
11710 ' リターンキー処理モードｺﾏﾝﾄﾞ処理
11720 '************************************
11730 *RETCLEAR.PROC
11740 INPUT "リターンキー処理を選択して下さい(OFF=0,改行=1,ｸﾘｱ=2) ";RETCLR%
11750 IF RETCLR%>2 THEN GOSUB *BEEP.ERROR: GOTO *RETCLEAR.PROC
11760 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(RETCLR%)
11770 RETURN
11800 '************************************
11810 ' 仮想画面サイズ設定ｺﾏﾝﾄﾞ処理
11820 '************************************
11830 *IMAGEAREA.PROC
11840 MSGX$="仮想画面の横方向文字数": MSGY$="仮想画面の縦方向行数": GOSUB *XY.INP
11850 SEND$=SEND$+CHR$(X%)+CHR$(Y%)
11860 RETURN
11900 '************************************
11910 ' 表示画面サイズ設定ｺﾏﾝﾄﾞ処理
11920 '************************************
11930 *DSPAREA.PROC

11940 MSGX$="表示画面の横方向文字数": MSGY$="表示画面の縦方向行数": GOSUB *XY.INPS
11950 SEND$=SEND$+CHR$(X%)+CHR$(Y%)
11960 RETURN
12000 '************************************
12010 ' カーソルロケートｺﾏﾝﾄﾞ処理
12020 '************************************
12030 *CURLOCATE.PROC
12040 MSGX$="カーソルを表示する": MSGY$=MSGX$: GOSUB *XY.INP
12050 SEND$=SEND$+CHR$(X%)+CHR$(Y%)
12060 RETURN
12100 '************************************
12110 ' 表示画面ロケートｺﾏﾝﾄﾞ処理
12120 '************************************
12130 *AREALOCATE.PROC
12140 MSGX$="表示する仮想画面の": MSGY$=MSGX$: GOSUB *XY.INP
12150 SEND$=SEND$+CHR$(X%)+CHR$(Y%)
12160 RETURN
12200 '************************************
12210 ' メッセージ表示ｺﾏﾝﾄﾞ処理
12220 '************************************
12230 *DATADSP.PROC
12240 QTY%=1: MAX%=&H7F: MSG$="表示するデータ数": GOSUB *HEX.INP
12250 QTY%=HEX%[1]: MAX%=&HFF: MSG$="ＬＣＤ文字フォントコード": GOSUB *HEX.INP
12260 SEND$=SEND$+CHR$(QTY%)
12270 FOR I=1 TO QTY%
12280 SEND$=SEND$+CHR$(HEX%[I])12290 NEXT I
12300 RETURN

12400 '************************************
12410 ' ユーザーキャラクタ設定ｺﾏﾝﾄﾞ処理
12420 '************************************
12430 *USERCHAR.PROC
12440 QTY%=1: MAX%=&H7: MSG$="ユーザ定義文字コード": GOSUB *HEX.INP
12450 SEND$=SEND$+CHR$(HEX%[1])12460 QTY%=7: MAX%=&HFF: MSG$="フォントパターン": GOSUB *HEX.INP
12470 FOR I=1 TO 712480 SEND$=SEND$+CHR$(HEX%[I])12490 NEXT I
12500 SEND$=SEND$+CHR$(0)12510 RETURN
12600 '************************************
12610 ' キーイニシャライズｺﾏﾝﾄﾞ処理
12620 '************************************
12630 *KEYINIT.PROC
12640 INPUT "キー初期化の対象を選択して下さい(両方=0,ｷｰｺﾝ=1,ｷｰﾏｸﾛ=2) ";KEYI%
12650 IF KEYI%>2 THEN GOSUB *BEEP.ERROR: GOTO *KEYINIT.PROC
12660 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(KEYI%)
12670 RETURN
12700 '************************************
12710 ' キーコンデータ設定ｺﾏﾝﾄﾞ処理
12720 '************************************
12730 *KEYCON.PROC
12740 QTY%=1: MAX%=&HFF: MSG$="登録テーブル情報": GOSUB *HEX.INP:MODE%=HEX%[1]12750 QTY%=1: MAX%=&H7F: MSG$="フルキーボードのキーナンバー": GOSUB*HEX.INP
12760 KEYNO%=HEX%[1]12770 MAX%=15: MSG$="登録文字数": GOSUB *QTY.INP
12780 MAX%=&HF4: MSG$="キーコード": GOSUB *HEX.INP

12790 SEND$=SEND$+CHR$(MODE%)+CHR$(KEYNO%)+CHR$(QTY%)
12800 FOR I=1 TO QTY%
12810 SEND$=SEND$+CHR$(HEX%[I])12820 NEXT I
12830 RETURN
12900 '************************************
12910 ' キーマクロ設定ｺﾏﾝﾄﾞ処理
12920 '************************************
12930 *KEYMACRO.PROC
12940 QTY%=1: MAX%=&H1D: MSG$="ＥＮＴＲＹ−１のキーナンバー": GOSUB*HEX.INP
12950 KEYNO%=HEX%[1]12960 MAX%=64: MSG$="登録文字数": GOSUB *QTY.INP
12970 MAX%=&HF4: MSG$="キーコード": GOSUB *HEX.INP
12980 SEND$=SEND$+CHR$(KEYNO%)+CHR$(QTY%)12990 FOR I=1 TO QTY%
13000 SEND$=SEND$+CHR$(HEX%[I])13010 NEXT I
13020 RETURN
13100 '************************************
13110 ' キーマスクモードｺﾏﾝﾄﾞ処理
13120 '************************************
13130 *KEYMASK.PROC
13140 QTY%=1: MAX%=&H7: MSG$="マスクのモード": GOSUB*HEX.INP
13150 MASKMODE%=HEX%[1]13160 IF MASKMODE%=2 OR MASKMODE%=3 OR MASKMODE%=6 OR MASKMODE%=7 THEN
*KEYMASK.ALL
13170 IF MASKMODE%<4 THEN MAX%=&H7F ELSE MAX%=&H1D
13180 MSG$="指定キーボードのキーナンバー": GOSUB *HEX.INP
13190 SEND$=SEND$+CHR$(MASKMODE%)+CHR$(HEX%[1])13200 RETURN
13210 *KEYMASK.ALL
13220 SEND$=SEND$+CHR$(MASKMODE%)
13230 RETURN
13300 '************************************
13310 ' キー配列選択ｺﾏﾝﾄﾞ処理
13320 '************************************
13330 *KEYMODE.PROC
13340 INPUT "フルキーボードのキー配列を選択して下さい(ﾉｰﾏﾙ=0,ｱｲｳｴｵ=1,ﾉﾝｼﾌﾄ=2,両方=3)";KEYMODE%
13350 IF KEYMODE%>3 THEN GOSUB *BEEP.ERROR: GOTO *KEYMODE.PROC
13360 GOSUB *BEEP.CLICK: SEND$=SEND$+CHR$(KEYMODE%)
13370 RETURN
13400 '************************************
13410 ' リピートキー設定ｺﾏﾝﾄﾞ処理
13420 '************************************
13430 *KEYREP.PROC
13440 INPUT "リピートのモードを選択して下さい(1ｷｰｵﾌ=0,1ｷｰｵﾝ=1,全ｷｰｵﾌ=2,全ｷｰｵﾝ=3) ";KEYREP%
13450 IF KEYREP%>3 THEN GOSUB *BEEP.ERROR: GOTO *KEYREP.PROC
13460 GOSUB *BEEP.CLICK
13470 IF KEYREP%>1 THEN GOTO *KEYREP.ALL
13480 QTY%=1: MAX%=&H1D: MSG$="エントリー１のキーナンバー": GOSUB*HEX.INP
13490 SEND$=SEND$+CHR$(KEYREP%)+CHR$(HEX%[1])13500 RETURN
13510 *KEYREP.ALL
13520 SEND$=SEND$+CHR$(KEYREP%)
13530 RETURN
13600 '************************************
13610 ' キーリピートインターバルｺﾏﾝﾄﾞ処理

13620 '************************************
13630 *KEYTIME.PROC
13640 QTY%=1: MAX%=&HFF: MSG$="インターバルタイム": GOSUB*HEX.INP
13650 SEND$=SEND$+CHR$(HEX%[1])13660 RETURN
13700 '************************************
13710 ' テストモード処理
13720 '************************************
13730 *TESTMODE.PROC
13740 ON STOP GOSUB *TESTMODE.END: STOP ON
13750 CLS: PRINT "ＥＮＴＲＹ−１からのＬＣＤキャラクタコードを表示します"
13760 PRINT "終了時は、本体キーボードのＳＴＯＰキーを押して下さい"
13770 TESTEND.FLG=OFFF
13780 WHILE( TESTEND.FLG=OFFF )13790 A$=INPUT$( 1 ): B$=RIGHT$("0"+HEX$(ASC(A$)),2)
13800 IF A$<" " THEN A$=" "13810 PRINT " ";A$;" [ ";B$;" ] ";13820 WEND13830 STOP OFF
13840 RETURN
13900 *TESTMODE.END
13910 TESTEND.FLG=ONN
13920 RETURN
50000 '************************************
50010 ' スタッククリア関数
50020 '************************************
50030 *STACK.CLEAR
50040 WHILE( INKEY$<>"" ): WEND

50050 RETURN
50100 '************************************
50110 ' クリック音を鳴らす関数
50120 '************************************
50130 *BEEP.CLICK
50140 BEEP 150150 FOR I.BEEP=1 TO 256: NEXT I.BEEP
50160 BEEP 050170 RETURN
50200 '************************************
50210 ' エラー音を鳴らす関数
50220 '************************************
50230 *BEEP.ERROR
50240 FOR I.BEEP=1 TO 250250 BEEP 150260 FOR J.BEEP=1 TO 777: NEXT J.BEEP
50270 BEEP 050280 FOR J.BEEP=1 TO 256: NEXT J.BEEP
50290 NEXT I.BEEP
50300 GOSUB *STACK.CLEAR
50310 RETURN
50400 '************************************
50410 ' １６進数入力関数
50420 '************************************
50430 *HEX.INP
50440 FOR I.HEX=1 TO QTY%
50450 *HEX.ERROR
50460 IF QTY%>1 THEN PRINT "No.";I.HEX;" : ";MSG$; ELSE PRINT MSG$;
50470 PRINT "(0〜";RIGHT$("0"+HEX$(MAX%),2);")";

50480 INPUT "を16進数にて入力して下さい ";H$50490 HEX%[I.HEX]=VAL("&H"+H$)
50500 IF HEX%[I.HEX]>MAX% THEN GOSUB *BEEP.ERROR: GOTO *HEX.ERROR
50510 IF HEX%[I.HEX]=0 AND LEFT$(H$,1)<>CHR$(&H30) THEN GOSUB *BEEP.ERROR: GOTO
*HEX.ERROR
50520 GOSUB *BEEP.CLICK
50530 NEXT I.HEX
50540 RETURN
50600 '************************************
50610 ' 数入力関数
50620 '************************************
50630 *QTY.INP
50640 PRINT MSG$;"(0〜";MAX%;")";: INPUT "を入力してください ";QTY%
50650 IF QTY%>MAX% THEN GOSUB *BEEP.ERROR: GOTO *QTY.INP
50660 GOSUB *BEEP.CLICK
50670 RETURN
50700 '************************************
50710 ' Ｘ座標・Ｙ座標入力関数
50720 '************************************
50730 *XY.INP
50740 PRINT MSGX$;: INPUT "Ｘ座標(0〜255)を入力してください ";X%50750 IF X%>255 THEN GOSUB *BEEP.ERROR: GOTO *XY.INP ELSE GOSUB *BEEP.CLICK
50760 *XY.ERROR
50770 PRINT MSGY$;: INPUT "Ｙ座標(0〜255)を入力してください ";Y%50780 IF Y%>255 THEN GOSUB *BEEP.ERROR: GOTO *XY.ERROR
50790 IF (X%+1)*(Y%+1)>2000 THEN GOSUB *BEEP.ERROR: GOTO *XY.INP
50800 GOSUB *BEEP.CLICK
50810 RETURN
50900 '************************************

50910 ' Ｘ座標・Ｙ座標入力関数(表示画面用)
50920 '************************************
50930 *XY.INPS
50940 PRINT MSGX$;: INPUT "Ｘ座標(0〜15)を入力してください ";X%50950 IF X%>15 THEN GOSUB *BEEP.ERROR: GOTO *XY.INPS ELSE GOSUB *BEEP.CLICK
50960 *XY.ERRORS
50970 PRINT MSGY$;: INPUT "Ｙ座標(0〜1)を入力してください ";Y%50980 IF Y%>1 THEN GOSUB *BEEP.ERROR: GOTO *XY.ERRORS
50990 IF (X%+1)*(Y%+1)>2000 THEN GOSUB *BEEP.ERROR: GOTO *XY.INPS
51000 GOSUB *BEEP.CLICK: RETURN
55000 '************************************
55010 ' ＥＮＴＲＹ−１にテータを送信する
55020 '************************************
55030 *SEND.DATA
55040 OUT &H2,(&H2 OR INP(&H2)) : DEF SEG=055050 SP.TABLE.1 =PEEK(&H538) : SP.TABLE.2 =PEEK(&H53A)
55060 TB.OFFSET.1=PEEK(&H522) : TB.OFFSET.2=PEEK(&H523)
55070 PRINT: PRINT "送信データ(16進)：";55080 FOR I.SEND=1 TO LEN(SEND$)
55090 OUT &H43,&H17:OUT &H41,ASC(MID$(SEND$,I.SEND,1)): A%=INP(&H41)
55100 PRINT RIGHT$("0"+HEX$(ASC(MID$(SEND$,I.SEND,1))),2);",";
55110 FOR J.SEND=1 TO 36: NEXT J.SEND
55120 NEXT I.SEND
55130 OUT &H43,&H1755140 POKE &H538,SP.TABLE.1 : POKE &H53A,SP.TABLE.2
55150 POKE &H522,TB.OFFSET.1 : POKE &H523,TB.OFFSET.2
55160 A%=INP(&H41): A%=INP(&H41)
55170 OUT &H2,(&HFD AND INP(&H2))
55180 PRINT CHR$(&H1D);" "55190 RETURN
60000 '************************************
60010 ' メニュー画面のデータ
60020 '************************************
60030 *MENU.DATA
60040 DATA 0, 2,4," *** 汎用コマンド *** "60050 DATA 27, 2,5," *** ＬＣＤコマンド *** "60060 DATA 55, 2,6," *** キーコマンド *** "60070 DATA 0, 4,4,"０：ENTRY-1イニシャライズ"60080 DATA 0, 5,4,"１：ブザー発鳴"
60090 DATA 0, 6,4,"２：キークリック音長設定"
60100 DATA 0, 7,4,"３：バックアップモード"
60110 DATA 0, 8,4,"４：ヘッダー設定"
60120 DATA 0, 9,4,"５：パケット転送モード"
60130 DATA 0,10,4,"６：デリミタ転送モード"
60140 DATA 0,11,4,"７：ディレイタイム設定"
60150 DATA 27, 4,5,"１０：LCDイニシャライズ"60160 DATA 27, 5,5,"１１：LCD表示状態設定"
60170 DATA 27, 6,5,"１２：LCD表示クリア"
60180 DATA 27, 7,5,"１３：カーソル種類設定"
60190 DATA 27, 8,5,"１４：カーソルトレースﾓｰﾄﾞ"
60200 DATA 27, 9,5,"１５：エコーバックモード"
60210 DATA 27,10,5,"１６：エコーバックﾃﾞｰﾀ変更"
60220 DATA 27,11,5,"１７：リターンキー処理ﾓｰﾄﾞ"
60230 DATA 27,12,5,"１８：仮想画面サイズ設定"
60240 DATA 27,13,5,"１９：表示画面サイズ設定"
60250 DATA 27,14,5,"１Ａ：カーソルロケート"
60266 DATA 27,15,5,"１Ｂ：表示画面ロケート"

60270 DATA 27,16,5,"１Ｃ：メッセージ表示"
60280 DATA 27,17,5,"１Ｄ：ユーザーｷｬﾗｸﾀ設定"
60290 DATA 55, 4,6,"２０：キーイニシャライズ"60300 DATA 55, 5,6,"２１：キーコンデータ設定"
60310 DATA 55, 6,6,"２２：キーマクロ設定"
60320 DATA 55, 7,6,"２３：キーマスクモード"
60330 DATA 55, 8,6,"２４：キー配列選択"
60340 DATA 55, 9,6,"２５：リピートキー設定"
60350 DATA 55,10,6,"２６：ｷｰﾘﾋﾟｰﾄｲﾝﾀｰﾊﾞﾙ"
60360 DATA 55,17,6,"３０：テストモード"
60370 DATA 0,17,4,"Ｅ：終了"

◆コマンドリファレンス・汎用コマンド
Ŷ００H エントリー１ イニシャライズ［機 能］ エントリー１を初期化します。
［コマンド］ １ＢH ００H ００H［解 説］ エントリー１に設定されている機能やデータを全て初期化します。
バックアップモードがＯＦＦの状態での、電源投入時と同じ状態で各機能は初期設定値になります。
機能グループ毎に初期化を行うことも可能です。
＜参 照＞ バックアップモード
ＬＣＤイニシャライズ
キーイニシャライズ
Ŷ０１H ブザー発鳴［機 能］ エントリー１の内蔵ブザーを鳴らします。
［コマンド］ １ＢH ０１H Ｐ１
Ｐ１・・・ブザー発鳴
００H ：ブザー発鳴停止
０１H〜ＦＥH：定時間ブザー発鳴
ＦＦH ：ブザー発鳴開始［解 説］
エントリー１内蔵のブザーを発鳴します。
またＰ１に０１HからＦＥHを設定することにより、ブザーを指定した時間だけ発鳴させることができます。
ブザー音長は、Ｐ１で指定し、最小０.０２秒から最大５.１０秒まで、０.０２秒単位で設定できます。
Ｐ１ ブザー発鳴時間
０１H・・・・・・・・・０.０２秒
０２H・・・・・・・・・０.０４秒
０３H・・・・・・・・・０.０６秒
・・・・
・・
ＦＥH・・・・・・・・・５.１０秒
ブザー音発生中にブザー停止コマンドを送信すると、ブザー音は停止します。
＜ご注意＞
ＦＦHのブザー発鳴開始は、００Hのブザー発鳴停止コマンド
またはエントリー１のキー入力があると停止します。
Ŷ０２H キークリック音長設定［機 能］
エントリー１キー押下時のクリック音長（ブザー音）を設定します。
［コマンド］ １ＢH ０２H Ｐ１

Ｐ１・・・クリック音長時間
００H ：クリック音無し
０１H〜ＦＦH：指定時間のクリック音［解 説］
エントリー１のキー押下時のクリック長を設定します。
Ｐ１の値によって無音から、最大５.１２秒まで設定可能です。
クリック音有りを設定する場合には、最小０.０２秒から最大５.１２秒まで０.０２秒毎に設定可能です。
Ｐ１ ブザー発鳴時間
０１H・・・・・・・・・０.０２秒
０２H・・・・・・・・・０.０４秒
０３H・・・・・・・・・０.０６秒
・・・・
・・
ＦＦH・・・・・・・・・５.１２秒＜初期値＞
初期設定値として０１H（０.０２秒）が設定されています。
Ŷ０３H バックアップモード［機 能］
エントリー１電源切断時における、ユーザー設定値のバックアップ（保持）状態を設定します。
［コマンド］ １ＢH ０３H Ｐ１
Ｐ１・・・バックアップモード
００H：ＯＦＦ（初期設定値を使用）
０１H：ＯＮ （ユーザー設定値を使用）［解 説]
本コマンドにより、使用設定状態やデータを保持します。
毎回使用するデータは、このコマンドを設定すると、電源切断時から約１０日間データをバックアップします。
保持内容は、ＬＣＤ仮想画面、表示画面、キーコンデータ、キーマクロデータ、ユーザーキャラク
タ、エコーバックデータです。
＜ご注意 ＞
バックアップ期間は８時間フル充電した時の目安であり、保証値ではありません。温度等の環境条
件や製品の老朽化により、バックアップ期間が短くなる場合もありますのでご注意ください。
本設定の変更は、ユーザーサービス機能で行なうことも可能ですが、ヘルプ機能で変更した場合
は、ヘルプ機能終了時から有効になります。
＜初期値＞
初期設定値として００H（バックアップＯＦＦ）が設定されています。
＜参 照 ＞ ユーザーサービス機能概要
Ŷ０４H ヘッダー設定［機 能］
送信データの先頭にヘッダー（識別子）を付加します。

［コマンド］ １ＢH ０４H Ｐ１ Ｄ１・・・・・Ｄｎ
Ｐ１・・・ヘッダーデータ数
００H：ヘッダー無し（ヘッダーデータの指定は必要ありません。）
０１H〜ＦＦH：
ヘッダーデータ長さ（バイト数）
Ｄ１・・・ヘッダーデータ
Ｐ１で指定した長さのデータを、
キーコードで指定してください。
＜ご注意 ＞
Ｐ１で００H（ヘッダー無し）を選択した場合はヘッダーデータを指定する必要はありません。
＜参 照＞ キーコード表［解 説］
パケット転送モードもしくは、デリミタ転送モードにおいて機種選択されている方からの、キー入力データに本ヘッダーを付加して、ＰＣ本体に転送します。
＜参 照＞ パケット転送モード
デリミタ転送モード＜初期値＞
初期設定値として００H（ヘッダー無し）が設定されています。
Ŷ０５H パケット転送モード［機 能］ 指定したデータサイズにてキーデータを送信します。
［コマンド］ １ＢH ０５H Ｐ１ Ｐ２
Ｐ１・・・機種選択
０１H：フルキーボード
０２H：エントリー１
０３H：両方混在
Ｐ２・・・パケットサイズ
００H：通常転送モード
パケット転送モードを解除し、通常転送モードにします。
０１H〜７ＦH：パケット転送モード
最大パケットデータサイズは７ＦHです。
［解 説]
フルキーボードもしくは、エントリー１から入力されたキーデータを指定したデータサイズまでバッファリング（蓄積）してから一度にＰＣに送信します。機種選択でフルキーボード、もしくはエントリー１を選択した場合に、もう一方は通常の転送状態になります。
両方混在を指定した場合は、指定されたパケットサイズになるまで、フルキーボードとエントリー
１の区別無しにキー入力データをバッファリングします。
＜ご注意＞1.フルキーボードとエントリー１に、それぞれ別のパケット
サイズを設定することはできません。

2.デリミタ転送モードとパケット転送モードの同時選択はできません。
＜初期値＞ 通常転送モードが設定されています。
Ŷ０６H デリミタ転送モード［機 能］ 指定したキーの押下でＰＣへデータを送信します。
［コマンド］ １ＢH ０６H Ｐ１ Ｐ２
Ｐ１・・・機種選択
０１H：フルキーボード
０２H：エントリー１
０３H：両方混在
Ｐ２・・・デリミタキーの指定＜参 照＞ キーコード表［解 説］
フルキーボードもしくは、エントリー１から入力されたキーデータをＰＣに送信するデリミタキー
を指定します。
デリミタキーが押されるまでのデータは、エントリー１内にバッファリング（蓄積）されます。機
種選択でフルキーボード、もしくはエントリー１を選択した場合に、もう一方は通常の転送状態になります。
両方混在を指定した場合は、いずれかのデリミタキー入力でデータ転送を行ないそれまではフルキーボードとエントリー１の区別無しに、キー入力データをバッファリングします。
本モードを解除して、通常の転送モードに切換えたい場合にはパケット転送モードのコマンドを利用して、Ｐ２のパケットサイズを００H （通常転送モード）にしてください。
＜ご注意＞
1.フルキーボードとエントリー１に、それぞれ別のデリミタを設定することはできません。
2.パケット転送モードとデリミタ転送モードの同時選択はできません。
Ŷ０７H ディレイタイム設定［機 能］
ＰＣへのデータ送信にディレイ（送信待ち時間）を入れます。
［コマンド］ １ＢH ０７H Ｐ１
Ｐ１・・・ディレイタイム
００H〜ＦＦH［解 説］
エントリー１から入力されたデータを、ＰＣ送信する各データ間に待ち時間を入れます。ＰＣに転送するデータ量が多い場合やＰＣでの処理に時間がかかる場合に利用します。ディレイタイムは、最小０秒（ディレイ無し）から最大５.１２秒まで、０.０２秒単位に設定できます。
Ｐ１ ディレイタイム
００H ・・・・・・・・０．００秒（ディレイ無し）
０１H ・・・・・・・・０．０２秒
０２H ・・・・・・・・０．０４秒
・・

・・
ＦＦH ・・・・・・・・５．１２秒
エントリー１では、キー入力をバッファリングしています。
バッファフルになるとキーボードからのデータ入力を無視します。
＜ご注意 ＞
デリミタ転送モードもしくはパケット転送モードで機種選択がされている場合には、選択されている機種からのキー入力データにのみディレイをかけて、ＰＣにデータを転送します。
＜初期値＞
初期設定値として００H（ディレイ無し）が設定されています。
コマンドリファレンス一覧表へ戻る

◆第４章 コマンドリファレンス・LCDコマンド
Ŷ１０H LCDイニシャライズ［機 能］ ＬＣＤの設定を初期設定状態にします。
［コマンド］ １ＢH １０H ００H［解 説]
ＬＣＤに関する全ての設定値を初期化します。
ユーザーの設定した、仮想画面サイズや表示画面サイズ等を初期設定値に戻し、ＬＣＤにはバックアップＯＦＦ時における電源投入時と同様に、下記のオープニングメッセージが表示されます。
ＥＮＴＲＹ−１ Ｖｅｒ×．××
Ｔｅｃｈｎｉｃａｌ Ｃｏｒｐ．

＜ご注意＞
ＬＣＤ表示内の×．××はバージョン表示ですので、製品のソフトウェア改良とともに数値が変わります。
＜参 照＞ ＬＣＤコマンドグループの各初期値
Ŷ１１H LCD表示状態設定［機 能］ ＬＣＤの表示状態を切換えます。
［コマンド］ １ＢH １１H Ｐ１
Ｐ１・・・表示設定
００H：表示ＯＦＦ
０１H：表示ＯＮ［解 説］
００H指定時には、ＬＣＤにデータを表示しません。
ＬＣＤ表示状態が、ＯＦＦの状態でも仮想画面のデータ書込は行われています。
＜初期値＞
初期設定値として０１H（表示ＯＮ）が設定されています。
Ŷ１２H LCD表示クリア［機 能］ ＬＣＤの仮想画面をクリアします。
［コマンド］ １ＢH １２H ００H［解 説］ 仮想画面全域をクリアします。
仮想画面をクリアし、表示画面及びカーソルは
ホームポジション（仮想画面左上）に戻ります。
＜参 照＞ 仮想画面サイズ設定
表示画面サイズ設定
Ŷ１３H カーソル種類設定［機 能］ ＬＣＤに表示するカーソルの状態を切換えます。
［コマンド］ １ＢH １３H Ｐ１ Ｐ２
Ｐ１・・・カーソル種類の選択

０１H：アンダーカーソル
０２H：ブリンクカーソル
０３H：アンダーブリンクカーソル
Ｐ２・・・表示状態
００H：カーソル表示ＯＦＦ
０１H：カーソル表示ＯＮ［解 説］
ＬＣＤにカーソルを表示します。
カーソルの種類は、アンダーカーソルとブリンクカーソルがあり同時表示も設定できます。
＜初期値＞ 初期設定値として以下の値が設定されています。
Ｐ１のカーソル種類は０１Hのアンダーカーソル
Ｐ２の表示状態は０１Hのカーソル表示ＯＮ
Ŷ１４H カーソルトレースモード［機 能］ カーソル移動にともなう表示画面のトレース（追従）モードを指定します。
［コマンド］ １ＢH １４H Ｐ１
Ｐ１・・・トレースモードの選択
００H：ＯＦＦ
０１H：ＯＮ［解 説］
仮想画面が表示画面より、大きく設定されている状態でカーソルトレースモードをＯＮにすると、カーソル表示が表示画面をはずれないように、カーソルの動きに合わせて、表示画面がトレースします。
［サンプル］
エントリー１にキーより１２３４５６７８９０１２３４５６７と入力した場合における表示画面の違いを示します。
（１）トレースモードＯＮの状態
３４５６７８９０１２３４５６７＿（２）トレースモードＯＦＦの状態
１２３４５６７８９０１２３４５６＊仮想画面上には表示されています。
＜初期値＞
初期設定値として０１H（追従する）が設定されています。
＜参 照＞ 仮想画面サイズ設定
Ŷ１５H エコーバックモード［機 能］
エントリー１キー入力のエコーバックモードを設定します。
［コマンド］ １ＢH １５H Ｐ１

Ｐ１・・・エコーバックモード
００H：ＯＦＦ
０１H：ＯＮ［解 説］
エントリー１のキー入力に対して、予め設定されているエコーバックデータをＬＣＤにエコーバッ
クします。
＜参 照＞ エコーバックデータ変更＜初期値＞
初期設定値として０１H（エコーバックモードＯＮ）が設定されています。
Ŷ１６H エコーバックデータ変更［機 能］
ＬＣＤへのキー入力エコーバックデータを変更します。
［コマンド］ １ＢH １６H Ｐ１ Ｐ２ Ｄ１・・・・・Ｄｎ
Ｐ１・・・エントリー１キー番号の指定
Ｐ２・・・エコーバックデータ数の指定
００H ：エコーバックしない
０１H〜０ＦH：エコーバックデータ数（最大１５文字まで）
Ｄ１・・・エコーバックデータ
００H 〜ＦＦH
Ｐ２のエコーバックデータ数で指定した分だけ、キャラクターコードを送ります。
＜ご注意＞
ＬＣＤキャラクタコード内で次のエリアは、機能コードに割当られています。
（１） ００H〜０１H（２）８０H〜９ＦH［解 説］
エコーバックモードＯＮにおいて、エントリー１キー押下時における、ＬＣＤへのエコーバックデ
ータを変更します。本データは、エントリー１本体上面のＬＣＤ表示に対してのみ行なわれるもの
で、ＰＣ本体に転送するキー入力データとは異なります。
またエコーバックデータに、機能コードを設定することにより、ユーザーサービス等が受けられるようになります。
コマンドリファレンス一覧表へ戻る

第４章 コマンドリファレンス・LCDコマンド
Ŷ１７H リターンキー処理モード［機 能］ リターンキー押下時のカーソル処理を選択します。
［コマンド］ １ＢH １７H Ｐ１
Ｐ１・・・モード選択
００H：ＯＦＦ（処理無し）
０１H：改行
０２H：クリア［解 説］
エントリー１のリターンキー押下時におけるカーソル処理を設定します。０２H のクリアでは仮想
画面全域をクリアして、カーソルはホームポジション（仮想画面左上）に戻ります。
＜初期値＞
初期設定値として０１H（改行）が設定されています。
Ŷ１８H 仮想画面サイズ設定［機 能］ 仮想画面の大きさを設定します。
［コマンド］ １ＢH １８H Ｐ１ Ｐ２
Ｐ１・・・Ｘ 横方向 １行の文字数
００H〜ＦＦH
Ｐ２・・・Ｙ 縦方向 行数
００H〜ＦＦH＜ご注意＞ Ｐ１×Ｐ２≦２０００文字［解 説］
仮想画面の文字数は最大で合計２０００文字です。
Ｘ＝００H、Ｙ＝００Hと指定すると、大きさとしては、１文字×１文字の仮想画面となり、仮想画
面がなくなるのではありません。
＜初期値＞
初期設定値として１６文字２行の仮想画面が設定されています。
Ｐ１は０ＦH（１６文字）
Ｐ２は０１H（２行）＜ご注意＞
本コマンドが実行されると、仮想画面の内容はクリアされ表示画面はホームポジション（仮想画面左上）になります。
＜参 照＞ 仮想画面とは
Ŷ１９H 表示画面サイズ設定［機 能］ 表示画面の大きさを設定します。
［コマンド］ １ＢH １９H Ｐ１ Ｐ２
Ｐ１・・・Ｘ 横方向 １行の文字数
HH

００ 〜０Ｆ
Ｐ２・・・Ｙ 縦方向 行数
００H〜０１H［解 説］
Ｘ＝００H、Ｙ＝００Hと指定すると、大きさとしては、１文字×１文字の表示画面となり、表示画
面がなくなるのではありません。
＜初期値＞
初期設定値として１６文字２行の表示画面が設定されています。
Ｐ１は０ＦH（１６文字）
Ｐ２は０１H（２行）＜参 照＞ 表示画面とは
Ŷ１AH カーソルロケート［機 能］ カーソルを指定した位置に移動します。
［コマンド］ １ＢH １ＡH Ｐ１ Ｐ２
Ｐ１・・・Ｘ 横方向
００H〜ＦＦH
Ｐ２・・・Ｙ 縦方向
００H〜ＦＦH［解 説］
Ｘ、Ｙは仮想画面中の位置を指定します。
但し、既に設定されている仮想画面のサイズを越えて指定することはできません。
＜参 照＞ 仮想画面サイズ設定
仮想画面とは
Ŷ１BH 表示画面ロケート［機 能］ 表示画面を移動します。
［コマンド］ １ＢH １ＢH Ｐ１ Ｐ２
Ｐ１・・・Ｘ 横方向
００H〜ＦＦH
Ｐ２・・・Ｙ 縦方向
００H〜ＦＦH［解 説］
仮想画面の中で表示画面をどの位置にセットするかを指定します。
Ｘ、Ｙは表示画面の左上を示します。
＜参 照＞ 仮想画面サイズ設定
表示画面サイズ設定
仮想画面とは表示画面とは
Ŷ１CH メッセージ表示

［機 能］ ＬＣＤにメッセージデータを表示します。
［コマンド］ １ＢH １ＣH Ｐ１ Ｄ１・・・・・Ｄｎ
Ｐ１・・・表示データ数
０１H〜７ＦH
Ｄ１・・・表示データ
ＬＣＤキャラクタコード表＜参 照＞ ＬＣＤキャラクタコード表［解 説］
エントリー１にＰＣ本体からメッセージデータを送りＬＣＤに表示します。表示データ数で設定した文字数分のみデータのキャラクタコードを表示します。
Ŷ１DH ユーザーキャラクタ設定［機 能］ ユーザーキャラクタを定義します。
［コマンド］ １ＢH １ＤH Ｐ１ Ｄ１・・・Ｄ８
Ｐ１・・・キャラクタコード番号
００H〜０７H
Ｄ１・・・フォントデータ８バイト［解 説］
ユーザーキャラクタ（定義文字）のフォントをセットします。
エントリー１にはＬＣＤ表示キャラクタコードの００H〜０７Hの８キャラクタはユーザーが自由に文字を定義できます。
コマンドリファレンス一覧表へ戻る

第４章 コマンドリファレンス・キーコマンド
Ŷ２０H キーイニシャライズ［機 能］ キー設定のイニシャライズ（初期化）をします。
［コマンド］ １ＢH ２０H Ｐ１
Ｐ１・・・キーデータ・テーブル
００H：両方初期化
０１H：キーコン初期化
０２H：キーマクロ初期化［解 説］
キーコンデータ、キーマクロのデータ・テーブルを初期化します。
＜参 照＞ キーコンデータ設定
キーマクロ設定
Ŷ２１H キーコンデータ［機 能］ キーコンデータの設定をします。
［コマンド］
１ＢH ２１H Ｐ１ Ｐ２ Ｐ３ Ｄ１・・・・Ｄ１５
Ｐ１・・・コンバートモード
Ｂｉｔ:
０
１
２

３

４

５

６

７

TABLE : ノーマル シフト CAPS CAPS+シフト カナ カナ+シフト GRPH CTRL

Ｐ２・・・トリガーキー番号
００H 〜７ＦH ：フルキーボードのキー番号
Ｐ３・・・コンバートデータサイズ
０１H 〜０ＦH ：最大１５文字
Ｄ１・・・コンバートデータ
Ｐ３で指定したサイズ分だけコンバートデータを送ります。
＜参 照＞ フルキーボード キー番号表
キーコード表［解 説］ フルキーボードのキーコンデータを設定します。
＜参 照＞ キーコンとは
キーコン・キーマクロ コマンド注意事項

Ŷ２２H キーマクロ設定［機 能］ キーマクロの設定をします。
［コマンド］ １ＢH ２２H Ｐ１ Ｐ２ Ｄ１・・・Ｄ６４

Ｐ１・・・キー番号
００H 〜１ＤH
エントリー１のキー番号
Ｐ２・・・データサイズ
最大６４文字Ｄ１・・・データ
Ｐ２で指定したデータ数分だけ
キーコードデータを送ります。
＜参 照＞ エントリー１ キー番号表［解 説］
エントリー１の３０キーの各キー毎にフルキーボードのキーマクロを設定します。
＜参 照＞ キーコン・キーマクロ コマンド注意事項

Ŷ２３H キーマスクモード［機 能］ キーボードの入力をマスク（無視）します。
［コマンド］ １ＢH ２３H Ｐ１ Ｐ２
Ｐ１・・・マスクモード
００H：フルキーボード 指定キー ＯＮ
０１H：フルキーボード 指定キー ＯＦＦ＊０２H：フルキーボード 全キー ＯＮ＊０３H：フルキーボード 全キー ＯＦＦ
０４H：エントリー１ 指定キー ＯＮ
０５H：エントリー１ 指定キー ＯＦＦ＊０６H：エントリー１ 全キー ＯＦＦ＊０７H：エントリー１ 全キー ＯＮ（＊印については、Ｐ２を指定する必要はありません）
Ｐ２・・・キー番号
００H 〜７ＦH＜参 照＞ エントリー１ キー番号表
フルキーボード キー番号表［解 説］
フルキーボード及びエントリー１の、各々のキー入力についてマスクできます。
＜初期値＞
初期設定値としてフルキーボード、エントリー１ともに全キーＯＦＦが設定されています。
Ŷ２４H フルキー配列選択［機 能］ キーボードの配列を選択します。
［コマンド］ １ＢH ２４H Ｐ１
Ｐ１・・・配列機種
００H：ＪＩＳ標準配列

０１H：あいうえお配列
０２H：ノンシフトモード
０３H：あいうえお配列＋ノンシフトモード［解 説］
フルキーボードのキー配列を上記より選択することが可能です。
将来のバージョンでは、ドボラーク配列や新しいローマ字配列等を提供する予定です。
＜初期値＞
初期設定値としてＰ１は、００H（ＪＩＳ標準配列）が設定されます。
＜ 参 照 ＞ あいうえお配列表

Ŷ２５H リピートキー設定［機 能] エントリー１ キーリピートの有無を設定します。
［コマンド］ １ＢH ２５H Ｐ１ Ｐ２
Ｐ１・・・キーモード
００H：１キーＯＦＦ
０１H：１キーＯＮ
０２H：全キーＯＦＦ
０３H：全キーＯＮ
Ｐ２・・・キー番号
００H〜１ＤH＜ご注意 ＞
Ｐ１で０２H もしくは０３H を設定した場合Ｐ２のキー番号は設定する必要はありません。
＜参照＞
エントリー１ キー番号表［解 説］ エントリー１の３０キーそれぞれについて、キーリピートの有無を設定します。
キー番号 リピート キー番号 リピート
００H 〜０９H ：ＯＦＦ ０ＤH 〜０ＥH ：ＯＦＦ
０ＡH 〜０ＣH ：ＯＮ ０ＦH 〜１ＤH ：ＯＮ＜参照＞
キーリピートインターバル設定＜初期値＞
初期設定値として各キー毎にリピートの有無が設定されています。
Ŷ２６H キーリピートインターバル［機 能］
エントリー１のキーリピートのインターバル（間隔）を設定します。
［コマンド］ １ＢH ２６H Ｐ１
Ｐ１・・・インターバルタイム値
００H〜ＦＦH［解 説］
ＰＣにキーデータを送信するデータ間の時間を設定します。
インターバルタイムはＰ１で指定し、最小０.０２秒から最大５.１２秒まで、０.０２秒単位で設定できます。
Ｐ１ インターバルタイム
０１H ・・・・・・・・・０．０２秒
０２H ・・・・・・・・・０．０４秒
０３H ・・・・・・・・・０．０６秒．．
．．
．．
ＦＦH ・・・・・・・・・５．１２秒＜参 照＞ リピートキー設定＜初期値＞
初期設定値としてＰ１は００H（標準インターバル）が設定されています。
コマンドリファレンス一覧表へ戻る